Image
Processing
IN IDL

IDL Version 6.2

July 2005 Edition
Copyright © RSI
All Rights Reserved



Restricted Rights Notice

The IDL®, ION Script™, and ION Java™ software programs and the accompanying procedures,
functions, and documentation described herein are sold under license agreement. Their use, dupli-
cation, and disclosure are subject to the restrictions stated in the license agreement. RSI reserves
the right to make changes to this document at any time and without notice.

Limitation of Warranty

RSI makes no warranties, either express or implied, as to any matter not expressly set forth in the
license agreement, including without limitation the condition of the software, merchantability, or
fitness for any particular purpose.

RSI shall not be liable for any direct, consequential, or other damages suffered by the Licensee or
any othersresulting from use of the IDL or ION software packages or their documentation.

Permission to Reproduce this Manual

If you are alicensed user of this product, RSI grantsyou alimited, nontransferable license to repro-
duce this particular document provided such copies are for your use only and are not sold or dis-
tributed to third parties. All such copies must contain the title page and this notice pagein their
entirety.

Acknowledgments

IDL® isaregistered trademark and ION™, |ON Script™, ION Java™, are trademarks of ITT Industries, registered in the United
States Patent and Trademark Office, for the computer program described herein.

Numerical Recipes™ is atrademark of Numerical Recipes Software. Numerical Recipes routines are used by permission.
GRG2™ isatrademark of Windward Technologies, Inc. The GRG2 software for nonlinear optimization is used by permission.

NCSA Hierarchical Data Format (HDF) Software Library and Utilities
Copyright 1988-2001 The Board of Trustees of the University of Illinois
All rights reserved.

NCSA HDF5 (Hierarchical Data Format 5) Software Library and Utilities
Copyright 1998-2002 by the Board of Trustees of the University of Illinois. All rights reserved.

CDF Library
Copyright © 2002 National Space Science Data Center
NASA/Goddard Space Flight Center

NetCDF Library
Copyright © 1993-1999 University Corporation for Atmospheric Research/Unidata

HDF EOS Library
Copyright © 1996 Hughes and Applied Research Corporation

This software is based in part on the work of the Independent JPEG Group.
Portions of this software are copyrighted by DataDirect Technologies, 1991-2003.

Portions of this software were developed using Unisearch's Kakadu software, for which Kodak has a commercial license. Kakadu
Software. Copyright © 2001. The University of New South Wales, UNSW, Sydney NSW 2052, Australia, and Unisearch Ltd,
Australia

Portions of this computer program are copyright © 1995-1999 LizardTech, Inc. All rightsreserved. MrSID is protected by U.S. Patent
No. 5,710,835. Foreign Patents Pending.

Portions of this software are copyrighted by Merge Technologies Incorporated.

This product includes software devel oped by the Apache Software Foundation (http://www.apache.org/)
IDL Wavelet Toolkit Copyright © 2002 Christopher Torrence.

Other trademarks and registered trademarks are the property of the respective trademark holders.



Contents

Chapter 1
Introduction to Image Processing in IDL ... 9
Overview of IMage PrOCESSING ......cocveieeierieeseeseeseeseesseesseesseessesssesssesssssssssssssesssesssenss 10
Digital Images and IMage ProCESSING .....cccvevrerieiieeieieseseeiese e se e et ene e 10
Understanding Image DefinitionSin IDL ......ccooceiiieiicceceece et 12
Representing IMage Datain IDL ......ccccceiiieiiee e 13
ACCESSING [MAGES ...oeeeeeeieeesee ettt et e e e e st e s tesae e s ae e s re e sreesreesneesreesree e 15
QUENYING IMAJES ...vveveitiieeeee sttt sttt st besre e e e testesbe e s e stesreereenaensesreas 15
REFEIEINCES ...ttt ettt e e re et e te s et e ne e e e stesneeneeneeseeneen 16
Chapter 2
Transforming Image GEOMELIY ...couviiiiiiiiiiiii e 17
Overview of Geometric TransformMations .........ccceoeeierereneneee e 18
CroppPiNg IMBOES .....ceiiriiieiriesieriee ettt sttt b e e b s b e e enis 20
[ (o [ o T =0 =SS 23
RESIZING IMBOES ....veeeeeiietieteee ettt ettt b e et b e sn e ens 26

Image Processing in IDL 3



ShIftiNG IMAJES ....veeeeeeece ettt b et esbesresneensentesreeneas 28
REVEISING IMBOES ..ottt r e et r e r et nn e e e 30
TranSPOSING IMBOES .....eevveiieeieeiesie et ete ettt e ste e re e e s e st st eesa e besbesae e e e testeeneeneeseesreenes 32
ROLALING IMAJES .....eeeeiriiteeee et n e e e nne s 34
Rotating an Image by 90 Degree INCrements ........c.ccevvveeeereneseeceesese e 34
Using the ROT Function for Arbitrary ROLaLIONS ........ccoceveeieiiieeeeee e 36
Planar Slicing of VOIUMELIHC DELA ........cceeeevieiiesiectieese s 38
Displaying a Series of Planar SHCES ......ccoeereiireiiere e 38
Extracting a Slice of VOIUMELIC Data .......cccocvevveveeevievecececese e 40
Interactive Planar Slicing of VOIUMELNC Data ........ccoovveeeereineneeere e 41
Displaying Volumetric DataUsing SLICERS .........cccccooeiieieve e 42
Manipulating Volumetric Data Using SLICERS .........ccoceiieieiiieeeeee e 43
Chapter 3
Mapping an Image onto GEOMELIY ....ccoeeieeeieeeiiiiieeeeeeerr e 47
Mapping Images onto SUIfaCES OVEIVIEW ........ccceceeverieeiee e sieeseeseeseesee e e eseeeseeenees 48
Mapping an Image onto EIevation Data ...........ccoeveeeireneneeieneseseeeeeses e 50
Opening Image and GEOMELY FIlES .......ccceviiiiececce e 50
Initializing the IDL Display ODJECES .......cceririirieirierie et 52
Displaying the Image and Geometric Surface ODJeCtS ......cccvvvvieevceeveerecre e, 53
Mapping an Image ONtO @ SPNEIE ..o e 57
Mapping an Image onto a Sphere Using Direct GraphicCs ..........ccccceveeveeveeneevenenne. 57
Mapping an Image onto a Sphere Using Object Graphics ..........cccoovievenecnieenneen. 60
Chapter 4
Working with Masks and Image StatistiCS ........cccvvvvvvvviiiiiiiiieeeeeee, 65
Overview of Masks and Image SEAtiSHICS .......ccveerveeeirerreieeeese e 66
MESKING IMBOES ....cviivieieeiiiecieeee et e st st sbe s s e et e s besbeeaeesaestesreesaesaeseesreennensens 68
(O TT ool g o N g =0 TSRS 72
Locating Pixel Vauesinan IMage .......cccceeveieieceeese ettt 76
Calculating IMage SEALISHICS ....oveeeeeeeeieeeeeere ettt sae e seesae e eneeneas 80
Chapter 5
Warping IMAGES ...oooiiiiiiiiiiiirii et e e 85
Overview of Warping IMagES ........ccouvireriririserieeecsie st sse e 86
Tipsfor Selecting Control POINES ........cccoveirirencerese e 87
Creating Transparent Image OVEXIAYS ........cccveieiiiieieese sttt sre e sre e 88

Contents Image Processing in IDL



Displaying Image Transparencies Using Direct Graphics .........cccceveveveveeciennennene 88
Displaying Image Transparencies Using Object GraphiCs ..........cccoereeerienencnnennns 88
Warping Images Using DireCt GraphiCs .........cccccieviiieieseseseesee st sre e 89
Chapter 6
Working with Regions of Interest (ROIS) ......cccccciviviiiiiiiiiiiiieeeeeeee, 99
Overview of WOorking With ROIS .......cooiiiiiiieeeesese e 100
Contrasting an ROI’'s Geometric Areaand Mask Area ........ccceeeeveeveevecseesienneeens 101
Defining RegIONS OF INTEIESE ........coiiiiieieieriesie et 103
Displaying ROI Objectsin a Direct Graphics Window .........ccccceveeveerecreeneeneeseeene, 105
Programmatically Defining ROIS ........c.ccriiiriieeneseseseeee s 109
L0 YTl =N o o o 113
Creating and Displaying @an ROl M@SK .......ccccciiirieierineneeeinese e 118
Testing an ROI for Point ContaiNMENL .........ccceoeeiieieeiie e 122
Creating a Surface Mesh of @n ROI GIOUP ......c.coeeriererieneeiniese e 125
Chapter 7
Transforming Between DOMAaINS .....cccccvviviiiiiiiiiiiiiieieeee e e e e e s 129
Overview of Transforming Between Image DOMAINS ........cccvvvereeierienereeneeiesesenneens 130
Transforming Between Domains With FFT ... 132
Transforming to the Frequency DOMaIN ........cccccooiviierieie e 132
Displaying Imagesin the Frequency DOMain ..........ccccccveveevenesecceesese e 136
Transforming from the Frequency DOMaIN .........ccocoverieieneninieee e 140
Removing NOise With the FFT ..ot 143
Transforming Between Domains With Wavelets .........ccooovveeieniniieeeecee e 148
Transforming to the Time-Frequency DOMaIN .........ccccccveveveveneereesie e 148
Displaying Imagesin the Time-Frequency DOmain .........ccccooeeveeeieneneesenn e 152
Transforming from the Time-Frequency Domain .........cccccecvvvveevenieciceesee e, 155
Removing Noise with the Wavelet Transform ........cccooeeoeveneneeeeneneneeere e 158
Transforming to and from the Hough and Radon Domains ...........cccceecevvvececceseseenne. 161
Transforming to the Hough and Radon Domains (Projecting) .......cccceeveeeerereenene. 162
Transforming from the Hough and Radon Domains (Backprojecting) .................. 165
Finding Straight Lines with the Hough Transform ..........cccccooeeieieneneneecese e 168
Color Density Contrasting with the Radon Transform ...........ccceceevvieveeceececenee, 174

Image Processing in IDL Contents



Chapter 8
Contrasting and Filtering .....ccccoovieeii i 179
Overview of Contrasting and FilteriNg .......cccocvieieeie i 180
BYLE-SCAIING ...ttt et 183
WOrking With HiStOGIraMS .......cc.ecieiiiiieceeese ettt st sre s 186
Equalizing With HiSLOQramS .........coeeieiiniieeee e 187
Adaptive Equalizing With HiStOgrams .........ccccccvveiieie v 190
LS g T 7o IF= I 0= o S 195
LOW PSS FIltEITNG ....ecveiiiiicecee sttt s s ne e ene s 196
High Pass FilTErNG ...ocoveieeeeeee e 199
Directional FiltEITNG .....ccviicieee s 203
Laplacian Filtering .....ccocooe oo 206
SMOGLhING 8N TMAGE ...veveeeee et e sttt s re e e e e tenteeneens 211
Smoothing With AVErage ValUES .........ceeeieii e 211
Smoothing With Median VEUES .........ccceeeiiiieceeece et 215
Sharpening @n IMAJE .......ooueeee et e et e 220
== o g To = [0 S 224
Enhancing Edges with the Roberts Operator ...........coceiiiieeeieneneeee e 224
Enhancing Edges with the Sobel Operator ...........ccooveceeiie e 226
S 0TV o o T 229
Windowing to REMOVE NOISE .....ccvcveeeieieiiesieeees ettt 229
Lee Filtering to REMOVE NOISE .....c.ooieiiiiieeee e e 233
Chapter 9
Extracting and Analyzing Shapes .......cccccceiiiiiiiiieecccee e, 237
Overview of Extracting and Analyzing Image Shapes ........ccccvcvvevvevveesieesceeseeseeneen, 238
Applying aMorphological Structuring Element to an Image ...........ccoceevvveecnneene. 238
Determining Structuring Element Shapes and SIZeS ........ccccceveeveevee e veesecseccieenens 241
Determining Intensity Values for Threshold and SIrefCh ... 243
Thresholding @n IMEOE .......eccveieeie e et re e e e 244
SIretChing @N TMAJE ......ooveieeeeeeeee e 245
Eroding and Dilating Image OLJECES .......ccceveeiieieesecrecre st 246
CharacteristiCS Of EFOSION ......cccveceieeeeiese e eeese s see st eseestesneenaeseeseeeneas 246
CharacteristicS Of DIlaLION ......cccceiiieieeee e 246
Applying Erosion and Dilaion ............cceeeerieeienenesieeeese s 246
Smoothing With MORPH_OPEN ......ccoiiiiieerieeeiereesie e 251

Contents Image Processing in IDL



Smoothing With MORPH_CLOSE .......cocooiiiirieerene et 254
Detecting Peaks Of BIrightNeSS ......cc.eoeirieeeeee e 257
Creating Image ObjeCt BOUNUANES .......ccocuveieveeiiesieeeese e 260
Selecting Specific IMage ObJECLS ......cveieiiiieeeree e 264
Detecting EAges of IMage ObJECES ......cccvvviieeiere et snea 269
Creating DiStanCe MaDS ......ooveeieiiieeeees ettt e et s ae e e neeseeenas 272
Thinning IMage OBJECLS .......ooviiiceee et 275
Combining Morphological OPErations ...........ccceeerereerere e 280
ANAlYZIiNG IMBGE SNAPES ......ccveiiecieeecte sttt s re e et ne e 285

Using LABEL_REGION to Extract Image Object Information ............cccceeveene. 285

Using CONTOUR to Extract Image Object Information ..........c.ccceevvevveceneveenne. 289
1o 1= S 293

Image Processing in IDL Contents






Chapter 1

Introduction to Image
Processing in IDL

This chapter describes the following topics:

Overview of Image Processing .......... 10 AccessingImages.........coovvvunnn. 15
Understanding Image DefinitionsinIDL .. 12 References ......................... 16
Representing Image DatainIDL ......... 13

10 Chapter 1: Introduction to Image Processing in IDL

Overview of Image Processing

Today, the medical industry, astronomy, physics, chemistry, forensics, remote
sensing, manufacturing, and defense are just some of the many fieldsthat rely upon
images to store, display, and provide information about the world around us. The
challenge to scientists, engineers and business peopleisto quickly extract valuable
information from raw image data. Thisis the primary purpose of image processing —
converting images to information.

This book explains how to process images using IDL (Interactive Data L anguage).
IDL isahigh-level programming language that contains an extensive library of
image processing and analysis routines. With IDL, you can quickly accessimage data
and begin investigating the best way to extract useful information.

Each chapter introduces image processing topics and includes information regarding
when one method may be preferred over another to enhance specific image features.
Numerous step-by-step examplesillustrate IDL’'s image processing and analysis
routines, allowing you to quickly understand how to get the desired results when
working with your own image data. This book is not intended to be a complete source
for image processing knowledge, an advanced image processing manual or an image
processing reference guide. This book is designed to teach people how to use IDL to
perform basic image processing, and does not assume that they are already expertsin
the field of image processing.

Digital Images and Image Processing

A digital image is composed of agrid of pixels and stored as an array. A single pixel
represents a value of either light intensity or color. Images are processed to obtain
information beyond what is apparent given the image's initial pixel values. Image
processing tasks can include any combination of the following:

Modifying the Image View — Transforming, translating, rotating and resizing
images are common tasks used to focus the viewer’s attention on a specific area of
the image. Chapter 2, “ Transforming Image Geometry” provides information on how
to precisely position imagesusing IDL.

Adding Dimensionality to Image Data — Some images provide more information
when they are placed on a polygon, surface, or geometric shape such as a sphere.
Chapter 3, “Mapping an Image onto Geometry” shows how to display images over
surfaces and geometric shapes.

Working with Masks and Calculating Statistics — Image processing uses some
fundamental mathematical methods to alter image arrays. These include masking,

Overview of Image Processing Image Processing in IDL

clipping, locating, and statistics. Chapter 4, “Working with Masks and Image
Statistics’ introduces these operations and provides examples of masking and
calculating image statistics.

Warping Images — Some data acquisition methods can introduce an unwanted
curvature into an image. |mage warping using control points can realign an image
along aregular grid or align two images captured from different perspectives. See
Chapter 5, “Warping Images’ for more information.

Specifying Regions of Interest (ROIs) — When processing an image, you may
want to concentrate on a specific region of interest (ROI). ROIs can be determined,
displayed, and analyzed within IDL as described in Chapter 6, “Working with
Regions of Interest (ROIs)”.

Manipulating Images in Various Domains — One of the most useful toolsin
image processing is the ability to transform an image from one domain to another.
Additional information can be derived from images displayed in frequency, time-
frequency, Hough, and Radon domains. Moreover, some complex processing tasks
are simpler within these domains. See Chapter 7, “ Transforming Between Domains’
for details.

Enhancing Contrast and Filtering — Contrasting and filtering provide the ability
to smooth, sharpen, enhance edges and reduce noise within images. See Chapter 8,
“Contrasting and Filtering” for details on manipulating contrast and applying filters
to highlight and extract specific image features.

Extracting and Analyzing Shapes — Morphological operations provide a means
of determining underlying image structures. Used in combination, these routines
provide the ability to highlight, extract, and analyze features within an image. See
Chapter 9, " Extracting and Analyzing Shapes’ for details.

Before processing images, it isimportant to understand how images are defined, how
image data is represented, and how images are accessed (imported and exported)
within IDL. These topics are described within the following sections of this chapter:

* “Understanding Image Definitionsin IDL” on page 12
* “Representing Image Datain IDL” on page 13
e “Accessing Images’ on page 15

Image Processing in IDL Overview of Image Processing



12 Chapter 1: Introduction to Image Processing in IDL

Understanding Image Definitions in IDL

An understanding of basic image definitions is necessary before proceeding with
image processing tasks. Some routines are specificaly designed for certain types of
images. Binary, grayscale, and indexed images are two-dimensional arrays, while
RGB images are three-dimensional arrays. In which group an image belongsis
determined by its contents and how it relates to its color information.

Within IDL, an image can be categorized as follows:

Image Type Descriptions

Binary Images Binary images contain only two values (off or on). The off
valueis usually azero and the on value is usualy aone. This
type of image is commonly used as a multiplier to mask
regions within another image.

Grayscale Images | Graysca eimages represent intensities. Pixelsrange from least
intense (black) to most intense (white). Pixel values usually
range from 0 to 255 or are scaled to this range when displayed.

Indexed Images | Instead of intensities, a pixel value within an indexed image
relates to a color value within a color lookup table. Since
indexed images reference color tables composed of up to 256
colors, the data values of these images are usually scaled to
range between 0 and 255.

RGB Images Within the three-dimensional array of an RGB image, two of
the dimensions specify the location of apixel within animage.
The other dimension specifies the color of each pixel The
color dimension always has asize of 3 and is composed of the
red, green, and blue color bands (channels) of the image.

Table 1-1: Image Definitions

Note
Grayscale and binary images can actually be treated as indexed images with an
associated grayscale color table.

Color information can also be represented in other forms, which are described in
“Color Systems’ in Chapter 8 of the Using IDL manual.

Understanding Image Definitions in IDL Image Processing in IDL

Representing Image Data in IDL

Pixel valuesin an image file can be stored in many different data types. IDL
maintains 15 different datatypes. The original datatype of animageisreflected in
IDL when importing the image, but the type can be converted once the imageis
stored in an IDL variable. The following types are commonly used for images:

* Byte— An 8-bit unsigned integer ranging in value from 0 to 255. Pixelsin
images are commonly represented as byte data.

e Unsigned Integer — A 16-bit unsigned integer ranging from 0 to 65535.
e Signed Integer — A 16-bit signed integer ranging from -32,768 to +32,767.

e Unsigned Longword Integer — A 32-bit unsigned integer ranging in value
from O to approximately four billion.

* Longword Integer — A 32-bit signed integer ranging in value from
approximately minus two billion to plus two billion.

e Hoating-point — A 32-bit, single-precision, floating-point number in the
range from -10% to 1038, with approximately 6 or 7 decimal places of
significance.

e Double-precision — A 64-bit, double-precision, floating-point number in the
range from -10%%8 to 103% with approximately 14 decimal places of
significance.

While pixel values are commonly stored in files as whole numbers, they are usually
converted to floating-point or double-precision data types prior to performing
numerical computations. See the examples section of “REFORM” in the IDL
Reference Guide manual and “ Calculating Image Statistics” in Chapter 4 for more
information.

IDL provides predefined routines to convert data from one type to another. These
routines are shown in the following table:

Function Description
BYTE Convert to byte
BYTSCL Scale data to range from O to 255 and then convert to byte
UINT Convert to 16-bit unsigned integer

Table 1-2: Some IDL Data Type Conversion Functions

Image Processing in IDL Representing Image Data in IDL

Function Description
FIX Convert to 16-bit integer, or optionally other type
ULONG Convert to 32-bit unsigned integer
LONG Convert to 32-hit integer
FLOAT Convert to floating-point
DOUBLE Convert to double-precision floating-point

Table 1-2: Some IDL Data Type Conversion Functions (Continued)

Representing Image Data in IDL Image Processing in IDL

Accessing Images

How an image isimported into IDL depends upon whether it is stored in an
unformatted binary file or acommon image file format. IDL can query and import
image data contained in the image file formats listed in “ Supported File Formats’ in
Chapter 1 of the Using IDL manual.

Note
IDL can also import and export images stored in scientific data formats, such HDF
and netCDF. For more information on these formats, see the Scientific Data
Formats manual.

See “Importing and Writing Datainto Variables’ in Chapter 6 of the Using IDL
manual for details on data accessin IDL. This chapter and the IDL Reference Guide
provide details on the file access routines used in examples in the following chapters.

Querying Images

Common image file formats contain standardized header information that can be
queried. IDL provides the QUERY _IMAGE function to return valuable information
about images stored in supported image file formats. For information on using
QUERY _IMAGE, see “Returning Image File Information” in Chapter 7 of the Using
IDL manual.

Image Processing in IDL Accessing Images

18 Chapter 2: Transforming Image Geometry

Overview of Geometric Transformations

Geometric image transformation functions use mathematical transformationsto crop,
pad, scale, rotate, transpose or otherwise alter an image array to produce a modified
view of an image. The transformations described in this chapter are linear
transformations. For a description of non-linear geometric transformations, see
Chapter 5, “Warping Images’.

When an image undergoes a geometric transformation, some or all of the pixels
within the source image are relocated from their original spatial coordinatesto anew
position in the output image. When arelocated pixel does not map directly onto the
center of apixel location, but falls somewhere in between the centers of pixel
locations, the pixel’s value is computed by sampling the values of the neighboring
pixels. Thisresampling, aso known as interpolation, affects the quality of the output
image. See “Interpolation Methods” in Chapter 8 of the Using IDL manual for more
information.

Note
In this book, Direct Graphics examples are provided by default. Object Graphics
examples are provided in cases where significantly different methods are required.

The following list introduces image processing tasks and associated IDL image
processing routines covered in this chapter.

Task Routine(s) Description
“Cropping SIZE Focuses attention on important image
Images’ on CURSOR features by creating arectangular region
page 20. of interest.
“Padding SIZE Creates aborder around the perimeter of
Images’ on an image for presentation or advanced
page 23. filtering purposes.
“Resizing CONGRID Enlarges or shrinks an image.
Images” on REBIN
page 26.

Table 2-1: Image Processing Tasks and Related
Image Processing Routines

Overview of Geometric Transformations Image Processing in IDL

Task Routine(s) Description
“Shifting SHIFT Shiftsimage pixel values along any
Images’ on image dimension.
page 28.
“Reversing REVERSE Reverses array elementsto flip an image
Images’ on horizontally or vertically.
page 30.
“Transposing TRANSPOSE Interchanges array dimensions, reflecting
Images’ on the image about a 45 degree line.
page 32.
“Rotating ROTATE Rotates an image to any orientation,
Images’ on ROT using 90 degree or arbitrary increments.
page 34.
“Planar Slicing | EXTRACT_SLICE | Displaysasingle dlice or a series of
of Volumetric | g |CER3 planar dicesin asingle window or
Data’ on interactively extracts planar slices of
page 38. XVOLUME volumetric data.
Table 2-1: Image Processing Tasks and Related
Image Processing Routines (Continued)
Note

This chapter uses data files from the IDL. examples/data directory. Two files,
data.txt and index. txt, contain descriptions of the files, including array sizes.

Image Processing in IDL

20 Chapter 2: Transforming Image Geometry
Cropping Images

Cropping an image extracts a rectangular region of interest from the original image.
This focuses the viewer’s attention on a specific portion of the image and discards
areas of the image that contain less useful information. Using image cropping in
conjunction with image magnification allows you to zoom in on a specific portion of
the image. This section describes how to exactly define the portion of the image you
wish to extract to create a cropped image. For information on how to magnify a
cropped image, see “Resizing Images’ on page 26.

Image cropping requiresapair of (X, y) coordinates that define the corners of the new,
cropped image. The following example extracts the African continent from an image
of the world. Complete the following steps for a detailed description of the process.

Example Code
See cropworld.pro inthe examples/doc/image subdirectory of the IDL
installation directory for code that duplicates this example.

1. Opentheworld imagefile, using ther, G, B arguments to obtain the image's
color information:

world = READ PNG (FILEPATH ('avhrr.png', $
SUBDIRECTORY = ['examples',6 'data'l), R, G, B)

2. Preparethe display device and load the color table with the red, green and blue
values retrieved from the image file in the previous step:

DEVICE, RETAIN = 2, DECOMPOSED = 0
TVLCT, R, G, B

3. Get the size of theimage and prepare the window display using the dimensions
returned by the SIZE command:

worldSize = SIZE (world, /DIMENSIONS)
WINDOW, 0, XSIZE = worldSize[0], YSIZE = worldSizel[1l]

4. Display theimage:
TV, world

In this example, we will crop theimage to display only the African continent
as shown in the following figure. Two sets of coordinates, (LeftLowX,
LeftLowY) and (RightTopX, RightTopY), will be used to create the new,
cropped image array.

Cropping Images Image Processing in IDL

(RightTopX, RightTopY)

(LeftLowX, LeftLowY)

Figure 2-1: Defining the Boundaries of the Cropped Image Array

In the following step, use the CURSOR function to define the boundaries of
the cropped image. The values returned by the CURSOR function will be
defined as the variables shown in the previous image.

Note
To crop an image without interactively defining the cursor position, you can
use the actual coordinates of the cropped image array in place of the
coordinate variables, (LeftLowX, LeftLowY) and (RightTopX, RightTopY).
See CropWorld.pro inthe examples/doc/image subdirectory of the
IDL installation directory for an example.

5. Usethe cursor function to define the lower-left corner of the cropped image by
entering the following line:

CURSOR, LeftLowX, LeftLowY, /DEVICE

The cursor changesto across hair symbol when it is positioned over the
graphics window. Click in the areato the left and below the African continent.

Image Processing in IDL Cropping Images

Cropping Images

Chapter 2: Transforming Image Geometry

Note
Thevauesfor LeftLowx and LeftLowy appearinthe IDLDE Variable
Watch window. Alternately, use PRINT, LeftLowX, LeftLowY to display
these values.

Define the upper-right corner of the cropped image. Enter the following line
and then click above and to the right of the African continent.

CURSOR, RightTopX, RightTopY, /DEVICE

Name the cropped image and define its array using the lower-left and upper-
right x and y variables:

africa = world[LeftLowX:RightTopX, LeftLowY:RightTopY]
Prepare awindow based on the size of the new array:

WINDOW, 2, XSIZE = (RightTopX - LeftLowX + 1), $
YSIZE = (RightTopY - LeftLowY + 1)

Display the cropped image:

TV, africa

Your image should appear similar to the following figure.

Figure 2-2: Result of the Cropped Image Example

Padding Images

Image padding introduces new pixels around the edges of an image. The border
provides space for annotations or acts as a boundary when using advanced filtering
techniques.

This exercise adds a 10-pixel border to |eft, right and bottom of the image and a 30-
pixel border at the top allowing space for annotation. The diagonal linesin the
following image represent the area that will be added to the original image. For an
example of padding an image, complete the following steps.

Example Code
See paddedimage . pro inthe examples/doc/image subdirectory of the IDL

installation directory for code that duplicates this example.

30 pixel pad

10 pixel pads

Figure 2-3: Diagonal Lines Indicate Padding

To add a border around the earth image, complete the following steps:
1. Opentheworld imagefile:

earth = READ PNG(FILEPATH('avhrr.png',6 $
SUBDIRECTORY = ['examples', 'data'l), R, G, B)

Prepare the display device:
DEVICE, DECOMPOSED = 0, RETAIN = 2

Load the color table with the red, green and blue values retrieved from the
image in step 1 and modify the color table so that the final index value of each
color band is the maximum color value (white):

TVLCT, R, G, B
maxColor = !D.TABLE_SIZE - 1
TVLCT, 255, 255, 255, maxColor

Get the size of the image by entering the following line:
earthSize = SIZE(earth, /DIMENSIONS)

Define the amount of padding you want to add to the image. This example
adds 10 pixelsto the right and left sides of the image equalling atotal of 20
pixels aong the x-axis. We also add 30 pixels to the top and 10 pixelsto the
bottom of the image for atotal of 40 pixels along the y-axis.

Using the REPLICATE syntax, Result = REPLICATE (Value, D11, ..., D8]),
create an array of the specified dimensions, and set Value equal to the byte
value of the final color index to make the white border:

paddedEarth = REPLICATE (BYTE (maxColor), earthSize[0] + 20, $
earthSize[1] + 40)

Note
The argument BYTE (maxColor) inthe previous line produces awhite
background only when white is designated as the final index value for the
red, green and blue bands of the color table you are using. As shown in step
3, this can be accomplished by setting each color component (of the color
table entry indexed by maxColor) to 255.

See “Graphic Display Essentials’ in Chapter 8 of the Using IDL manual for
detailed information about modifying color tables.

6. Copy the original image, earth, into the appropriate portion of the padded

array. The following line places the lower-left corner of the original image
array at the coordinates (10, 10) of the padded array:

paddedEarth [10,10] = earth

7. Prepare awindow to display the image using the size of the original image plus
the amount of padding added along the x and y axes:

WINDOW, 0, XSIZE = earthSize[0] + 20, $
YSIZE = earthSize[l] + 40

8. Display the padded image.
TV, paddedEarth
9. Placeatitle at the top of the image using the XYOUTS procedure.

x = (earthSize[0]/2) + 10

y = earthSize[1l] + 15

XYOUTS, x, y, 'World Map', ALIGNMENT = 0.5, COLOR = 0, $
/DEVICE

The resulting image should appear similar to the following figure.

Werld Map

Figure 2-4: Resulting Padded Image

Resizing Images

Image resizing, or scaling, supports further image analysis by either shrinking or
expanding an image. Both the CONGRID and the REBIN functions resize one-, two-
or three-dimensional arrays. The CONGRID function resizes an image array by any
arbitrary amount. The REBIN function requires that the output dimensions of the new
array be an integer multiple of the original image’s dimensions.

When magnifying an image, new values are interpolated from the source image to
produce additional pixelsin the output image.When shrinking an image, pixels are
resampled to produce a lower number of pixels in the output image. The default
interpolation method varies according to whether you are magnifying or shrinking
the image.

When magnifying an image:

¢ CONGRID defaults to nearest-neighbor sampling with 1D or 2D arrays and
automatically uses bilinear interpolation with 3D arrays.

¢ REBIN defaults to bilinear interpolation.
When shrinking an image:
¢ CONGRID uses nearest-neighbor interpolation to resample the image.

* REBIN averages neighboring pixel valuesin the source image that contribute
to asingle pixel value in the output image.

The following example uses CONGRID since it offers more flexibility. However, if
you wish to resize an array proportionally, REBIN returns results more quickly. For
an example of magnifying an image using the CONGRID function, complete the
following steps.

Example Code
Seemagnifyimage.pro inthe examples/doc/image subdirectory of the IDL
installation directory for code that duplicates this example.

1. Select thefile and read in the data, specifying known data dimensions:

file = FILEPATH('convec.dat',6 $
SUBDIRECTORY = ['examples',6 'data'])
image = READ_BINARY(file, DATA_DIMS = [248, 248])

2. Load acolor table and prepare the display device:

LOADCT, 28
DEVICE, DECOMPOSED = 0, RETAIN = 2

3. Prepare the window and display the original image:

WINDOW, 0, XSIZE = 248, YSIZE = 248
TV, image

4. Usethe CONGRID function to increase the image array size to 600 by 600
pixels and force bilinear interpolation:

magnifiedImg = CONGRID (image, 600, 600, /INTERP)
5. Display the magnified image in a new window:
WINDOW, 1, XSIZE = 600, YSIZE = 600
TV, magnifiedImg
The following figure displays the original image (left) and the magnified view of the
image (right).

Figure 2-5: Original Image and Magnified Image

Image Processing in IDL Resizing Images



28 Chapter 2: Transforming Image Geometry
Shifting Images

The SHIFT function moves elements of avector or array along any dimension by any
number of elements. All shifts are circular. Elements shifted off one end are wrapped
around, appearing at the opposite end of the vector or array.

Occasionally, image files are saved with array elements offset. The SHIFT function
alows you to easily correct such images assuming you know the amounts of the
vertical and horizontal offsets. In the following example, the x-axis of original image
is offset by a quarter of the image width, and the y-axis is offset by athird of the
height.

Figure 2-6: Example of Misaligned Image Array Elements

Using the SHIFT syntax, Result = SHIFT (Array, S;, ..., Sp),wewill enter
negative values for the s (dimension) amounts in order to correct the image offset.

Example Code
See shiftimageoffset.prointheexamples/doc/image subdirectory of the
IDL installation directory for code that duplicates this example.

1. Select theimagefile and read it into memory:

file = FILEPATH('shifted endocell.png', $
SUBDIRECTORY = ['examples', 'data'l)
image = READ PNG(file, R, G, B)

2. Prepare the display device and load the image's associated color table:

DEVICE, DECOMPOSED = 0, RETAIN = 2
TVLCT, R, G, B
3. Get the size of theimage, prepare awindow based upon the values returned by
the SIZE function, and display the image to be corrected:

imageSize = SIZE(image, /DIMENSIONS)
WINDOW, 0, XSIZE = imageSize[0], YSIZE = imageSize[1l], $
TITLE = 'Original Image'
TV, image
4. Use SHIFT to correct the original image. Move the elements along the x-axis

to the left, using a quarter of the array width as the x-dimension values. Move
the y-axis elements, using one third of the array height as the number of
elements to be shifted. By entering negative values for the amount the image
dimensions are to be shifted, the array elements move toward the x and y axes.

image = SHIFT (image, - (imageSize[0]/4), - (imageSize[l1l]/3))
5. Display the corrected image in a second window:

WINDOW, 1, XSIZE = imageSize[0], YSIZE = imageSizel[l], $
TITLE='Shifted Image'
TV, image

The following figure displays the corrected image.

Figure 2-7: Resulting Shifted Array

Reversing Images

The REVERSE function allows you to reverse any dimension of an array. This
allows you to quickly change the viewing orientation of an image (flipping it
horizontally or verticaly).

Note that in the REVERSE syntax,
Result = REVERSE (Array [, Subscript Index] [, /OVERWRITE])

Subscript Index Specifiesthe dimension number beginning with 1, not O aswith
some other functions.

The following example demonstrates reversing the x-axis values (dimension 1) and
the y-axis values (dimension 2) of an image of aknee.

Example Code
Seereverseimage.pro inthe examples/doc/image subdirectory of the IDL
installation directory for code that duplicates this example.

1. Select the DICOM image of the knee and get the image’s dimensions:

image = READ DICOM (FILEPATH('mr knee.dcm', $
SUBDIRECTORY = ['examples', 'data'l))
imgSize = SIZE (image, /DIMENSIONS)

2. Prepare the display device and load the gray scale color table:

DEVICE, DECOMPOSED = 0, RETAIN = 2
LOADCT, O

3. Usethe REVERSE function to reverse the x-axis values (£ 1ipHorzImg) and
y-axisvalues (f1ipVertImg):

flipHorzImg = REVERSE (image, 1)
flipVertImg = REVERSE (image, 2)

4. Create an output window that is 2 times the size of the x-dimension of the
image and 2 times the size of the y-dimension of the image:

WINDOW, 0, XSIZE = 2*imgSize[0], YSIZE = 2*imgSize[l], $
TITLE = 'Original (Top) & Flipped Images (Bottom)''

5. Display theimages, controlling their placement in the graphics window by
using the Position argument to the TV command:

TV, image, O
TV, flipHorzImg, 2
TV, flipVertImg, 3

Your output should appear similar to the following figure.

Figure 2-8: Original Image (Top); Reversed Dimension 1 (Bottom Left); and
Reversed Dimension 2 (Bottom Right)

Transposing Images

Transposing an image array interchanges array dimensions, reflecting an image about
adiagonal (for example, reflecting a square image about a 45 degree line). By
default, the TRANSPOSE function reverses the order of the dimensions. However,
you can control how the dimensions are atered by specifying the optional vector, B,
in the following statement:

Result = TRANSPOSE (Arrayl[, P])

The valuesfor p start at zero and correspond to the dimensions of the array. The
following example transposes a photomicrograph of smooth muscle cells.

Example Code
See transposeimage.pro inthe examples/doc/image subdirectory of the
IDL installation directory for code that duplicates this example.

1. Openthefileand prepareto display it with acolor table:

READ JPEG, FILEPATH('muscle.jpg', $

SUBDIRECTORY=['examples', 'data'l), image
DEVICE, DECOMPOSED = 0, RETAIN = 2
LOADCT, O

2. Display the original image:
WINDOW, 0, XSIZE = 652, YSIZE = 444, TITLE = 'Original Image'
TV, image

3. Reducetheimage size for display purposes:

smallImg = CONGRID(image, 183, 111)

4. Using the TRANSPOSE function, reverse the array dimensions. This
essentially flips the image across its main diagonal axis, moving the upper left
corner of the image to the lower right corner.

transposeImgl = TRANSPOSE (smallImg)

WINDOW, 1, XSIZE = 600, YSIZE = 183, TITLE = 'Transposed
Images'

TV, transposeImgl, O

5. Specifying the reversal of the array dimensions leads to the same result since
thisisthe default behavior of the TRANSPOSE function.

transposeImg2 = TRANSPOSE (smallImg, [1,0])
TV, transposelmg2, 2

Transposing Images Image Processing in IDL


RSI_PROCODE/examples/doc/image/transposeimage.pro

Chapter 2: Transforming Image Geometry 33

6. However, specifying the original arrangement of the array dimensions results
in no image transposition.

transposeImg3 = TRANSPOSE (smallImg, [0,1]1)
TV, transposeImg3, 2

The following figure displays the original image (top) and the results of the various
TRANSPOSE statements (bottom).

Figure 2-9: Original (Top) and Transposed Images (Bottom) from Left to Right,
transposelmgl, transposelmg2, and transposelmg3

Rotating Images

To change the orientation of animagein IDL, use either the ROTATE or the ROT
function. The ROTATE function changes the orientation of an image by 90 degree
increments and/or transposes the array. The ROT function rotates an image by any
amount and offers additional resizing options. For more information, see “Using the
ROT Function for Arbitrary Rotations” on page 36.

Rotating an Image by 90 Degree Increments

The following example changes the orientation of an image by rotating it 270°.

Example Code
See rotateimage.pro inthe examples/doc/image subdirectory of the IDL
installation directory for code that duplicates this example.

1. Select thefile and read in the data, specifying known data dimensions:

file = FILEPATH('galaxy.dat',K $
SUBDIRECTORY=['examples', 'data'l)
image = READ BINARY (file, DATA DIMS = [256, 256])

2. Prepare the display device, load acolor table, create awindow, and display the
image:
DEVICE, DECOMPOSED = 0, RETAIN = 2
LOADCT, 4
WINDOW, 0, XSIZE = 256, YSIZE = 256
TVSCL, image
3. Using the ROTATE syntax, Result = ROTATE (Array, Direction), rotate the
galaxy image 270° counterclockwise by setting the Direction argument equal
to 3. See “ROTATE Direction Argument Options’ on page 35 for more
information.

rotateImg = ROTATE (image, 3)
4. Display the rotated image.

Window, 1, XSIZE = 256, YSIZE = 256,
TVSCL, rotatelImg

The following figure displays the original (left) and the rotated image (right).

Figure 2-10: Using ROTATE to Alter Image Orientation

ROTATE Direction Argument Options

The following table describes the Direction options available with the ROTATE
function syntax, Result = ROTATE (Array, Direction).

Direction Transpose? Coun?grtgrci)ocrllwise Slrirgglee

0 No None T—

1 No 90° -

> No 180° 1

3 No 270° T
4 Yes None -T

c Yes 90° -
6 Yes 180° l_

. Yes 270° -

Table 2-2: Direction Options Available with ROTATE

Using the ROT Function for Arbitrary Rotations

The ROT function supports clockwise rotation of an image by any specified amount
(not limited to 90 degree increments). Keywords also provide a means of optionally
magnifying the image, selecting the pivot point around which the image rotates, and
using either bilinear or cubic interpolation. If you wish to rotate an image only by 90
degree increments, ROTATE produces faster results.

The following example opens aimage of awhirlpool galaxy, rotates it 33° clockwise
and shrinksit to 50% of its original size.

Example Code
Seearbitraryrotation.prointheexamples/doc/image subdirectory of the
IDL installation directory for code that duplicates this example.

1. Select thefile and read in the data, specifying known data dimensions:

file = FILEPATH('m51.dat', $
SUBDIRECTORY = ['examples',6 'data'])
image = READ BINARY (file, DATA DIMS = [340, 440])

2. Prepare the display device and load a black and white color table:

DEVICE, DECOMPOSED = 0, RETAIN = 2
LOADCT, O

3. Create awindow and display the original image:

WINDOW, 0, XSIZE = 340, YSIZE = 440
TVSCL, image

4. Using the ROT function syntax,

Result=ROT (A, Angle, [Mag, X,, Y,] [,/INTERP]
[,CUBIC=value{-1 to 0}] [, MISSING=value] [,/PIVOT])

enter the following line to rotate the image 33°, shrink it to 50% of its original
size, and fill the image display with a neutral gray color where there are no
original pixel values:

arbitraryImg = ROT(image, 33, .5, /INTERP, MISSING = 127)

5. Display the rotated image in a new window by entering the following two
lines:

WINDOW, 1, XSIZE = 340, YSIZE = 440
TVSCL, arbitraryImg

Your output should appear similar to the following figure.

Figure 2-11: The Original Image (Left) and Modified Image (Right)

TheM1ssiNg keyword maintains the original image’s boundaries, keeping the
interpolation from extending beyond the original image size. Replacing MISSING =
127 with MISSING = 0 inthe previous example creates a black background by
using the default pixel color value of 0. Removing the mrssING keyword from the

same statement allows the image interpolation to extend beyond the image’s original
boundaries.

Planar Slicing of Volumetric Data

Volumetric displays are composed of a series of 2D dices of datawhich are layered
to produce the volume. IDL provides routines that allow you to display a series of the
2D dlicesin asingle image window, display single orthogonal or non-orthogonal
dlices of volumetric data, or interactively extract slices from a 3D volume. For more
information, see the following sections:

» “Displaying a Series of Planar Slices’ in the following section

e “Extracting a Slice of Volumetric Data’ on page 40

e “Interactive Planar Slicing of Volumetric Data” on page 41
Displaying a Series of Planar Slices

The following example displays 57 Magnetic Resonance Imaging (MRI) slices of a
human head within asingle window aswell asasingle slicewhich is perpendicular to
the MRI data.

Example Code
Seedisplayslices.prointheexamples/doc/image subdirectory of the IDL
installation directory for code that duplicates this example.

1. Select thefile and read in the data, specifying known data dimensions:

file = FILEPATH('head.dat', SUBDIRECTORY = ['examples',
'data']l)
image = READ BINARY (file, DATA DIMS = [80, 100, 57])

2. Load acolor table to more easily distinguish between data values and prepare
the display device:

LOADCT, 5
DEVICE, DECOMPOSED = 0, RETAIN = 2

3. Create the display window. When displaying all 57 slices of thearray in a
single window, the image size (80 by 100) and the number of dices (57)
determine the window size. In this case, 10 columnsand 6 rowswill contain all
57 dlices of the volumetric data.

WINDOW, 0, XSIZE = 800, YSIZE = 600

4. Usethevariable i inthe following FOR statement to incrementally display
each imagein the array. The i also functionsto control the positioning which,
by default, uses the upper left corner asthe starting point. Use 255b - array

to display the images using the inverse of the selected color table and the
ORDER keyword to draw each image from the top down instead of the bottom

up.
FOR 1 = 0, 56,1 DO TVSCL, 255b - image [*,*,i], /ORDER, i

5. To extract acentral slice from they, z plane, which is perpendicular to the x, y
plane of the MRI scans, specify 40 for the x-dimension value. Use REFORM
to decrease the number of array dimensions so that TV can display the image:

sliceImg = REFORM (image [40,*,*])
Thisresultsin a100 by 57 array.
6. Use CONGRID to compensate for the sampling rate of the scan dlices:
sliceImg = CONGRID(sliceImg, 100, 100)
7. Display the slicein the 47th window position:
TVSCL, 255b - sliceImg, 47

Since theimage size is now 100 x 100 pixels, the 47th position in the 800 by
600 window isthe final position.

Your output should be similar to the following figure.

Figure 2-12: Planar Slices of a MRI Scan of a Human Head

Note
This method of extracting slices of dataislimited to orthogonal slices only. You
can extract single orthogonal and non-orthogonal slices of volumetric data using
EXTRACT_SLICE, described in the following section. See “Extracting a Slice of
Volumetric Data” below for more information.

Extracting a Slice of Volumetric Data

The EXTRACT _SLICE function extracts a single two-dimensiona planar slice of
data from athree-dimensional volume. By setting arguments that specify the
orientation of the slice and a point in its center using the following syntax, you can
precisely control the orientation of the slicing plane.

Result = EXTRACT SLICE( Vol, Xsize, Ysize, Xcenter, Ycenter,
Zcenter, Xrot, Yrot, Zrot [, ANISOTROPY=[xspacing, yspacing,
zspacing]] [, OUT VAL=value] [, /RADIANS] [, /SAMPLE]

[, VERTICES=variable] )

The following example demonstrates how to use EXTRACT_SLICE to extract the
same singular dice as that shown in the previous example.

Example Code
Seeextractslice.prointhe examples/doc/image subdirectory of the IDL
installation directory for code that duplicates this example.

1. Select thefile and read in the data, specifying known data dimensions:

file = FILEPATH('head.dat', SUBDIRECTORY = ['examples',
'data'l])
volume = READ_BINARY(file, DATA DIMS =[80, 100, 57])

2. Prepare the display device and load the grayscale color table.

DEVICE, DECOMPOSED = 0, RETAIN = 2
LOADCT, O

3. Enter the following line to extract a sagittal planar dlice from the MRI volume
of the head.

sliceImg = EXTRACT SLICE $
(volume, 110, 110, 40, 50, 28, 90.0, 90.0, 0.0, OUT_VAL=0)

Note
The code within the previous parentheses specifies: the volume (pata), a
Size greater than the Xsize and Ysize of the volume (110, 110), the Xcenter,
Ycenter and Zcenter (40, 50, 28) denoting the x, y, and zindex points
through which the slice will pass, the degree of x, y, and z rotation of the
dlicing plane (90.0, 90.0, 0.0)andtheouT VAL = o indicating that
elements of the output array which fall outside the original values will be
given the value of O or black.

4. Use CONGRID to resize the output array to an easily viewable size. Thisis
also used to compensate for the sampling rate of the scan images.

bigImg = CONGRID (sliceImg, 400, 650, /INTERP)
5. Prepare adisplay window based on the resized array and display the image.

WINDOW, 0, XSIZE = 400, YSIZE = 650
TVSCL, bigImg

The image created by this example should appear similar to the following figure.

Figure 2-13: Example of Extracting a Slice of Data From a Volume

Interactive Planar Slicing of Volumetric Data

The series of two-dimensional images created by the magnetic resonance imaging
scan, shown in the section, “Displaying a Series of Planar Slices’ on page 38, can

also be visualized as a three-dimensional volume using either of IDL’s interactive
volume visualization tools, SLICER3 or XVOLUME.

SLICERS3 quickly creates visualizations of 3D datausing IDL Direct Graphics. The
XVOLUME procedure employs IDL Object Graphicsto create highly interactive
visualizations that take advantage of OpenGL hardware acceleration and multiple
processors for volume rendering. Since Object Graphics are rendered in memory and
not simply drawn, both the time and amount of virtual memory required to create a
XVOLUME visualization exceed those needed to create a Direct Graphics, SLICERS3
visualization.
Tip
For more information and examples of displaying volumes and dlicing volumetric
data using XVOLUME, see “XVOLUME” in the IDL Reference Guide manual.

Displaying Volumetric Data Using SLICERS3

The Direct Graphics SLICER3 widget-based application allows you to view single or
multiple dlices of avolume or to create an isosurface of the three-dimensional data.
Complete the following steps to load the head . dat volumeinto the SLICER3
application.

Example Code
Seedisplayslicer3.prointhe examples/doc/image subdirectory of the
IDL installation directory for code that duplicates this example.

1. Select the datafile and read in the data using known dimensions:

file = FILEPATH('head.dat',6 $
SUBDIRECTORY=['examples', 'data'])
volume = READ_BINARY(file, DATA DIMS = [80, 100, 571)

2. Todisplay al slices of the head . dat fileasavolumein SLICERS, create a
pointer called pdata which passesthe data array information to the
SLICERS3 application.

pbata = PTR_NEW (volume)

Note

You can load multiple arrays into the SLICERS3 application by creating a pointer for
each array. Each array must have the same dimensions.

3. Loadthe datainto the SLICER3 application. The DATA_NAMES designates
the data set in the application’s Data list. Thisfield will be greyed out if only
one volumetric array has been loaded.

SLICER3, pData, DATA NAMES ='head'

At first it is not apparent that your data has been passed to the SLICER3 application.
See the following section, “Manipulating Volumetric Data Using SLICER3” for
details on how to use thisinterface.

Manipulating Volumetric Data Using SLICERS3

Once you have loaded a three-dimensional array into the SLICERS3 application, the

interface offers numerous ways to visualize the data. The following steps cover

creating an isosurface, viewing a slice of data within the volume and rotating the

display.

1. Inthe SLICERS application, select Surface from the Mode: list. Left-click in

the Surface Threshold window containing the logarithmic histogram plot of
the data and drag the green line to change the threshold value of the display. A
value in the low to mid 40'sworks well for thisimage. Click Display to view
the isosurface of the data.

Fie Took About

Surface Thieshokl

41.443262

Figure 2-14: An Isosurface of Volumetric Data

Note
To undo an action resulting in an unwanted image in the SLICERS3 window,
you can either choose Tools — Delete and select the last item on the list to
undo the last action or choose Tools — Erase to erase the entire image.

2. Select Slice from the Mode list. Select the Expose, Orthogonal, and X
options. Left-click in the image window and drag the mouse halfway along the
X axis and then rel ease the mouse button. The planar slice of volumetric data
appears at the point where you release the mouse button.

Fie Tools About

Mode:[Sice =]

 Dwawe % Expose

& Outhogonal
" Oblque

Fx CY C2

Figure 2-15: Visualizing a Slice of Volumetric Data

3. Change the colors used to display the slice by selecting Tools — Colors —
Slice/Block. Inthe color table widget, select STD Gamma-11 from thelist and
click Done to load the new color table.

4. Changethe view of the display by selecting View from the M ode list. Here
you can change the rotation and zoom factors of the displayed image. Use the
dlider barsto rotate the orientation cube. A preview of the cube’s orientation
appearsin the small window above the controls. To create the orientation
shown in the following figure, move the slider to arotation of -18 for Z and -80
for X. Click Display to change the orientation of the image in the window.

The following figure displays the final image.

Figure 2-16: A Slice Overlaying an Isosurface

To save the image currently in the display window, select File —» Save — Save TIFF
Image. For more information about using the SLICER3 interface to manipulate
volumetric data, see “SLICER3” in the IDL Reference Guide.

Note
Enter the following line after closing the SLICER3 application to release memory

used by the pointer: PTR_FREE, pData

Mapping an Image onto
Geometry

This chapter describes the following topics:

Mapping Images onto Surfaces Overview .. 48  Mapping an Image onto aSphere . .. ... .. 57
Mapping an Image onto Elevation Data ... 50

Image Processing in IDL 47



48 Chapter 3: Mapping an Image onto Geometry

Mapping Images onto Surfaces Overview

Mapping an image onto geometry, also known as texture mapping, involves
overlaying an image or function onto a geometric surface. Images may be redlistic,
such as satellite images, or representational, such as col or-coded functions of
temperature or elevation. Unlike volume visualizations, which render each voxel
(volume element) of a three-dimensional scene, mapping an image onto geometry
efficiently creates the appearance of complexity by simply layering an image onto a
surface. The resulting realism of the display also provides information that is not as
readily apparent as with asimple display of either the image or the geometric surface.

Mapping an image onto a geometric surface is atwo step process. First, theimageis
mapped onto the geometric surface in object space. Second, the surface undergoes
view transformations (relating to the viewpoint of the observer) and is then displayed
in 2D screen space. You can use IDL Direct Graphics or Object Graphics to display
images mapped onto geometric surfaces.

The following table introduces the tasks and routines covered in this chapter.

Task Routine(s)/Object(s) Description
“Mapping an SHADE _SURF Display the elevation data.
Image_onto , IDLgrwWindow::Init Initialize the objects necessary
Elevation Data Lo . S
on page 50 IDLgrView::Init for an Object Graphics display.

' IDLgrModel::Init
IDLgrSurface::Init Initialize a surface object
containing the elevation data.
IDLgrImage::Init Initialize an image object
containing the satellite image.
XOBIVIEW Display the object in an

interactive IDL utility allowing
rotation and resizing.

Table 3-1: Tasks and Routines Associated with Mapping an Image onto
Geometry

Mapping Images onto Surfaces Overview Image Processing in IDL

Task Routine(s)/Object(s) Description
“Mapping an MESH_OBJ Create a sphere.
Image onto a REPLICATE
SphereUsing "o g Specify system variables
Direct Graphics required for 3D viewin
on page 57. equi viewing.
SET_SHADING Contral the light source used by
POLY SHADE.
TVSCL Map the image onto the sphere
POLY SHADE using POLY SHADE and
display the example with
TVSCL.
“Mapping an MESH_OBJ Create a sphere.
Image onto a REPLICATE
SphereUsmg. ., | IDLgrModel::Init Initialize model, palette and
Object Graphics . . )
on page 60 IDLgrPalette::| ryt image objects.
' IDLgrImage::Init
FINDGEN Create normalized coordinates
REPLICATE in order to map the image onto

the sphere.

IDLgrPolygon::Init

Assign the sphere to a polygon
object and apply the image
object.

XOBJIVIEW

Display the object in an
interactive IDL utility allowing
rotation and resizing.

Table 3-1: Tasks and Routines Associated with Mapping an Image onto
Geometry (Continued)

Image Processing in IDL

Mapping an Image onto Elevation Data

The following Object Graphics example maps a satellite image from the Los
Angeles, Californiavicinity onto aDEM (Digital Elevation Model) containing the
ared stopographical features. The realism resulting from mapping the image onto the
corresponding elevation data provides a more informative view of the area’s
topography. The processis segmented into the following three sections:

*  “Opening Image and Geometry Files’, in the following section

e “Initializing the IDL Display Objects’ on page 52

« “Displaying the Image and Geometric Surface Objects’ on page 53

Note
Data can be either regularly gridded (defined by a 2D array) or irregularly gridded
(defined by irregular X, y, z points). Both the image and elevation data used in this
example are regularly gridded. If you are dealing with irregularly gridded data, use
GRIDDATA to map the data to aregular grid.

Complete the following steps for a detailed description of the process.

Example Code
Seeelevation object.pro intheexamples/doc/image subdirectory of the
IDL installation directory for code that duplicates this example.

Opening Image and Geometry Files

The following steps read in the satellite image and DEM files and display the
elevation data.

1. Select the satellite image:

imageFile = FILEPATH('elev_t.jpg', $
SUBDIRECTORY = ['examples',6 'data'])

2. Import the JPEG file:
READ JPEG, imageFile, image
3. Select the DEM file:

demFile = FILEPATH('elevbin.dat',6 $
SUBDIRECTORY = ['examples', 'data'l])

Mapping an Image onto Elevation Data Image Processing in IDL

4. Define an array for the elevation data, open the file, read in the data and close
thefile:

dem = READ BINARY (demFile, DATA DIMS = [64, 64])

5. Enlargethe size of the elevation array for display purposes.
dem = CONGRID (dem, 128, 128, /INTERP)

6. To quickly visualize the elevation data before continuing on to the Object
Graphics section, initialize the display, create awindow and display the
elevation data using the SHADE_SURF command:

DEVICE, DECOMPOSED = 0

WINDOW, 0, TITLE = 'Elevation Data'
SHADE SURF, dem

: Elevation Data

Figure 3-1: Visual Display of the Elevation Data

After reading in the satellite image and DEM data, continue with the next section to
create the objects necessary to map the satellite image onto the el evation surface.

Image Processing in IDL Mapping an Image onto Elevation Data

Initializing the IDL Display Objects

After reading in the image and surface data in the previous steps, you will need to
create objects containing the data. When creating an IDL Object Graphics display, it
is hecessary to create awindow object (oWindow), aview object (oView) and a model
object (oModel). These display objects, shown in the conceptual representation in the
following figure, will contain a geometric surface object (the DEM data) and an
image object (the satellite image). These user-defined objects are instances of
existing IDL object classes and provide access to the properties and methods
associated with each object class.

L g—OWindow - an IDLgrWindow object

oView - an IDLgrView object
oMode - an IDLgrModel object

\ oSurface - the geometric elevation object

olmage - the satellite image object

Figure 3-2: Conceptualization of Object Graphics Display Example

Note
The XOBJVIEW utility (described in “Mapping an Image onto a Sphere Using
Object Graphics’ on page 60) automatically creates window and view objects.

Complete the following steps to initialize the necessary IDL objects.

1. [Initialize the window, view and model display objects. For detailed syntax,
arguments and keywords available with each object initialization, see
IDLgrWindow::Init, IDLgrView::Init and IDLgrModel::Init. The following
three lines use the basic syntax oNewObject = OBJ NEW ('Class Name')
to create these abjects:

oWindow = OBJ NEW ('IDLgrWindow', RETAIN = 2, COLOR MODEL = 0)
oView = OBJ NEW('IDLgrView')
oModel = OBJ_NEW ('IDLgrModel"')

Mapping an Image onto Elevation Data Image Processing in IDL

2. Assign the elevation surface data, dem, to an IDLgrSurface object. The

IDLgrSurface::Init keyword, STYLE = 2, drawsthe elevation datausing a
filled line style:

oSurface = OBJ NEW('IDLgrSurface', dem, STYLE = 2)

3. Assign the satellite image to a user-defined | DL grlmage object using
IDLgrImage::Init:

oImage = OBJ NEW('IDLgrImage',6 image, INTERLEAVE = 0, $
/INTERPOLATE)

INTERLEAVE = 0 indicatesthat the satellite image is organized using pixel
interleaving, and therefore has the dimensions (3, m, n). The INTERPOLATE
keyword forces bilinear interpolation instead of using the default nearest-
neighbor interpolation method.

Displaying the Image and Geometric Surface Objects

This section displays the objects created in the previous steps. The image and surface

objects will first be displayed in an IDL Object Graphics window and then with the
interactive XOBJVIEW tility.

1. Center the elevation surface object in the display window. The default object
graphics coordinate system is[—1,—1], [1,1]. To center the object in the
window, position the lower left corner of the surface data at [-0.5,-0.5, —0.5]
for the x, y and z dimensions:

oSurface -> GETPROPERTY, XRANGE = xr, YRANGE = yr, $
ZRANGE = zr

xs = NORM_COORD (xr)

xs[0] = xs[0] - 0.5

ys = NORM_COORD (yr)

ys[0] = ys[0] - 0.5

zs = NORM_COORD (zr)

zs[0] = zs[0] - 0.5

oSurface -> SETPROPERTY, XCOORD CONV = xs, $
YCOORD _CONV = ys, ZCOORD = zs

2. Map the satellite image onto the geometric el evation surface using the
IDLgrSurface::Init TEXTURE_MAP keyword:

oSurface -> SetProperty, TEXTURE MAP = oImage, $
COLOR = [255, 255, 255]

For clearest display of the texture map, set COLOR = [255, 255, 255]. If the
image does not have dimensions that are exact powers of 2, IDL resamplesthe
image into alarger size that has dimensions which are the next powers of two

Image Processing in IDL Mapping an Image onto Elevation Data

greater than the original dimensions. This resampling may cause unwanted
sampling artifacts. In this example, the image does have dimensions that are
exact powers of two, so no resampling occurs.

If your texture does not have dimensions that are exact powers of 2 and you
do not want to introduce resampling artifacts, you can pad the texture with
unused data to a power of two and tell IDL to map only a subset of the
texture onto the surface.

For example, if your imageis 40 by 40, create a 64 by 64 image and fill part
of it with the image data:

textureImage = BYTARR (64, 64)
textureImage [0:39, 0:39] = image ; image is 40 by 40
oImage = OBJ NEW('IDLgrImage', textureImage)

Then, construct texture coordinates that map the active part of the textureto a
surface (oSurface):

textureCoords = [[], [1, [1, []1
oSurface -> SetProperty, TEXTURE COORD = textureCoords

The surface object in IDL 5.6 is has been enhanced to automatically perform
the above calculation. In the above example, just use the image data (the 40
by 40 array) to create the image texture and do not supply texture
coordinates. IDL computes the appropriate texture coordinates to correctly
use the 40 by 40 image.

Note

Some graphic devices have alimit for the maximum texture size. If your
textureislarger than the maximum size, IDL scalesit down into dimensions
that work on the device. Thisrescaling may introduce resampling artifacts
and loss of detail in the texture. To avoid this, usethe TEXTURE_HIGHRES
keyword to tell IDL to draw the surface in smaller pieces that can be texture
mapped without loss of detail.

3. Add the surface object, covered by the satellite image, to the model object.

Then add the model to the view object:

oModel -> Add, oSurface
oView -> Add, oModel

Mapping an Image onto Elevation Data Image Processing in IDL

4. Rotate the model for better display in the object window. Without rotating the
model, the surface is displayed at a 90° elevation angle, containing no depth
information. The following lines rotate the model 90° away from the viewer
along the x-axis and 30° clockwise along the y-axis and the x-axis:

oModel -> ROTATE, [1, 0, 0], -90
oModel -> ROTATE, [0, 1, 0], 30
oModel -> ROTATE, [1, 0, 0], 30

5. Display the result in the Object Graphics window:

oWindow -> Draw, oView

Figure 3-3: Image Mapped onto a Surface in an Object Graphics Window

6. Display theresultsusing XOBJVIEW, setting the scaLe = 1 (instead of the
default value of 1/SQRT3) to increase the size of the initial display:
XOBJVIEW, oModel, /BLOCK, SCALE = 1

Thisresultsin the following display.

Figure 3-4: Displaying the Image Mapped onto the Surface in XOBJVIEW

After displaying the model, you can rotate it by clicking in the application
window and dragging your mouse. Select the magnify button, then click near
the middle of theimage. Drag your mouse away from the center of the display
to magnify the image or toward the center of the display to shrink the image.
Select the left-most button on the XOBJVIEW toolbar to reset the display.

7. Destroy unneeded object references after closing the display windows:

OBJ_DESTROY, [oView, oImage]

The oModel and oSurface objects are automatically destroyed when oMew is
destroyed.

For an example of mapping an image onto a regular surface using both Direct and
Object Graphics displays, see “Mapping an Image onto a Sphere” on page 57.

Mapping an Image onto a Sphere

The following example maps an image containing a color representation of world
elevation onto a sphere using both Direct and Object Graphics displays. The example
is broken down into two sections:

¢ “Mapping an Image onto a Sphere Using Direct Graphics’
«  “Mapping an Image onto a Sphere Using Object Graphics’ on page 60

Mapping an Image onto a Sphere Using Direct
Graphics

Complete the following steps for a detailed description of the process.

Example Code
Seemaponsphere direct.pro intheexamples/doc/image subdirectory of
the IDL installation directory for code that duplicates this example.

1. Select the file containing the world elevation image. Define the array, read in
the data and close the file:

file = FILEPATH('worldelv.dat',6 $
SUBDIRECTORY = ['examples',6 'data'])
image = READ_BINARY(file, DATA_DIMS = [360, 360])

2. Prepare the display device to display a PseudoColor image:
DEVICE, DECOMPOSED = 0

3. Load acolor tableand using TVLCT, set the final index value of thered, green
and blue bands to 255 (white). Setting these index values to white provides for
the creation of awhite window background in alater step.

LOADCT, 33
TVLCT, 255,255,255, !D.TABLE.SIZE - 1

(For comparison, TvLCT, 0, 0, 0, !D.TABLE SIZE+1 would designate
ablack window background.)

4. Create awindow and display the image containing the world elevation data:

WINDOW, 0, XSIZE = 360, YSIZE = 360
TVSCL, image

Image Processing in IDL Mapping an Image onto a Sphere

Thisimage, shown in the following figure, will be mapped onto the sphere.

Figure 3-5: World Elevation Image

5. Use MESH_OBJto create a sphere onto which the image will be mapped. The
following line specifies avalue of 4, indicating a spherical surface type:

MESH OBJ, 4, vertices, polygons, REPLICATE(0.25, 360, 360), $
/CLOSED

The vertices and polygons variables are the lists that contain the mesh vertices
and mesh indices of the sphere. REPLICATE generates a 360 by 360 array,
each element of which will contain the value 0.25. Using REPLICATE in the
Arrayl argument of MESH_OBJ specifies that the vertices variableisto
consist of 360 by 360 vertices, each positioned at a constant radius of 0.25
from the center of the sphere.

6. Create awindow and define the 3D view. Use SCALE3 to designate
transformation and scaling parameters for 3D viewing. The AX and AZ
keywords specify the rotation, in degrees about the x and z axes:

WINDOW, 1, XSIZE = 512, YSIZE = 512

SCALE3, XRANGE = [-0.25,0.25], YRANGE = [-0.25,0.25], 8
ZRANGE = [-0.25,0.25], AX = 0, AZ = -90

7. Set thelight sourceto contral the shading used by the POLY SHADE function.
Use SET_SHADING to modify the light source, moving it from the default
position of [0,0,1] with rays parallel to the z-axis to alight source position of
[-0.5, 0.5, 2.0]:

SET SHADING, LIGHT = [-0.5, 0.5, 2.0]

8. Set the system background color to the default color index, defining awhite
window background:
!P.BACKGROUND = !P.COLOR
9. Use TVSCL to display the world elevation image mapped onto the sphere.
POLY SHADE references the sphere created with the MESH_OBJroutine, sets

SHADES = image to map the image onto the sphere and uses the image
transformation defined by the T3D transformation matrix:

TVSCL, POLYSHADE (vertices, polygons, SHADES = image, /T3D)

The specified view of the image mapped onto the sphere is displayed in a
Direct Graphics window as shown in the following figure.

Figure 3-6: Direct Graphics Display of an Image Mapped onto a Sphere

10. After displaying the image, restore the system’s default background color:
! P.BACKGROUND = 0

To create a Object Graphics display featuring a sphere that can be interactively
rotated and resized, compl ete the steps contained in the section, “Mapping an Image
onto a Sphere Using Object Graphics’ below.

Mapping an Image onto a Sphere Using Object
Graphics

This example maps an image containing world elevation data onto the surface of a
sphere and displays the result using the XOBJVIEW utility. Thisutility automatically
creates the window object and the view object, previously shown in the section,
“Initializing the IDL Display Objects’ on page 52. Therefore, this example creates an
object based on IDLgrModel that contains the sphere, the image and the image
palette, as shown in the conceptual representation in the following figure.

4«— oM odel - an IDLgrModel object
containing the sphere, image, and
palette

oPolygon - an object defining the sphere,
containing the image and palette

ol mage - an object containing the image
oPalette - an object defining the color table

Figure 3-7: Conceptualization of XOBJVIEW Object Graphics Example

Complete the following steps for a detailed description of the process.

Example Code
Seemaponsphere object.pro intheexamples/doc/image subdirectory of
the IDL installation directory for code that duplicates this example.

Note
If you are continuing the exercise from the previous section, “Mapping an Image
onto a Sphere Using Direct Graphics’, skip steps 1, and 2. Proceed with step 3 to
create the necessary objects.

1. Select the world elevation image. Define the array, read in the data and close

thefile.
file = FILEPATH('worldelv.dat',6 S
SUBDIRECTORY = ['examples',6 'data'])
image = READ BINARY (file, DATA DIMS = [360, 360])

2. Usethe MESH_OBJ procedureto create a sphere onto which theimage will be
mapped. The following invocation of MESH_OBJ uses avalue of 4, which
represents a spherical mesh:

MESH OBJ, 4, vertices, polygons, REPLICATE(0.25, 101, 101)

When the MESH_OBJ procedure compl etes, the vertices and polygons
variables contain the mesh vertices and polygonal mesh connectivity
information, respectively. Although our image is 360 by 360, we can texture
map the image to a mesh that has fewer vertices. IDL interpolates the image
data across the mesh, retaining all the image detail between polygon vertices.
The number of mesh vertices determines how close to perfectly round the
sphere will be. Fewer vertices produce a sphere with larger facets, while more
vertices make a sphere with smaller facets and more closely approximates a
perfect sphere. A large number of mesh verticeswill increase the time required
to draw the sphere. In this example, MESH_OBJ produces a 101 by 101 array
of verticesthat are located in a sphere shape with aradius of 0.25.

3. Initialize the display objects. In this example, it is necessary to define a model
object that will contain the sphere, the image and the color table palette. Using
the syntax, oNewObject = OBJ NEW ('Class Name'), Create the model,
pal ette and image objects:

oModel = OBJ NEW ('IDLgrModel')

oPalette = OBJ NEW('IDLgrPalette')

oPalette -> LOADCT, 33

oPalette -> SetRGB, 255, 255, 255, 255

oImage = OBJ NEW('IDLgrImage',6 image, PALETTE = oPalette)

The previous lines initialize the oPalette object with the color table and then
set the final index value of the red, green and blue bands to 255 (white) in
order to use white (instead of black) to designate the highest areas of elevation.
The palette object is created before the image object so that the palette can be
applied when initializing the image object. For more information, see
IDLgrModel::Init, IDLgrPalette::Init and IDLgrlmage::Init.

4. Create texture coordinates that define how the texture map is applied to the
mesh. A texture coordinate is associated with each vertex in the mesh. The
value of the texture coordinate at a vertex determines what part of the texture
will be mapped to the mesh at that vertex. Texture coordinates run from 0.0 to
1.0 across atexture, so atexture coordinate of [0.5, 0.5] at a vertex specifies
that the image pixel at the exact center of the image is mapped to the mesh at
that vertex.

In this example, we want to do a simple linear mapping of the texture around
the sphere, so we create a convenience vector that describes the mapping in

each of the texture’s x- and y-directions, and then create these texture
coordinates:

vector = FINDGEN(101)/100.

texure coordinates = FLTARR(2, 101, 101)

texure coordinates[0, *, *] vector # REPLICATE(1., 101)
texure coordinates([1l, *, *] REPLICATE (1., 101) # wvector

The code above copies the convenience vector through the array in each
direction.

Enter the following line to initialize a polygon object with the image and
geometry data using the IDLgrPolygon::Init function. Set suaDING = 1 for
gouraud (smoother) shading. Set the paTa keyword equal to the sphere defined
with the MESH_OBJ function. Set coLor to draw awhite sphere onto which
the image will be mapped. Set TEXTURE COORD equal to the texture
coordinates created in the previous steps. Assign the image object to the
polygon object using the TEXTURE_MaP keyword and force bilinear
interpolation:

oPolygons = OBJ NEW('IDLgrPolygon', SHADING = 1, $
DATA = vertices, POLYGONS = polygons, $
COLOR = [255, 255, 2551, $
TEXTURE COORD = texure coordinates, $
TEXTURE _MAP = oImage, /TEXTURE INTERP)

Note
When mapping an image onto an IDLgrPolygon object, you must specify
both TEXTURE_MAP and TEXTURE_COORD keywords.

Add the polygon containing the image and the palette to the model object:
oModel -> ADD, oPolygons
Rotate the model -90° along the x-axis and y-axis:

oModel -> ROTATE, [1, 0, 0], -90
oModel -> ROTATE, [0, 1, 0], -90

Display the results using XOBJVIEW, an interactive utility allowing you to
rotate and resize objects:

XOBJVIEW, oModel, /BLOCK

After displaying the object, you can rotate the sphere by clicking in the display
window and dragging your mouse. Select the magnify button and click near
the middle of the sphere. Drag your mouse away from the center of the display
to magnify the image or toward the center of the display to shrink the image.

Select the left-most button on the XOBJVIEW toolbar to reset the display. The
following figure shows a rotated and magnified view of the world elevation
object.

&l Xobjview M= E3
File Edit “iew

[ o] [#[4] i |

=

Figure 3-8: Magnified View of World Elevation Object

9. After closing the XOBJVIEW display, remove unneeded object references:

OBJ_DESTROY, [oModel, oImage, oPalette]

Chapter 4

Working with Masks
and Image Statistics

This chapter describes the following topics:

Overview of Masks and Image Statistics. .. 66 Locating Pixel ValuesinanImage . ...... 76
Maskinglmages...................... 68 Calculating Image Statistics ............ 80
Clippinglmages...................... 72

Overview of Masks and Image Statistics

Mathematical operations used with imagesinclude logic (conditional) operations and
statistics. Logic operations are used to make masks to apply threshold levelsto clip
the pixel values of an image, and to locate pixel values. These operations help to
segment featuresin an image, after which statistics can be derived to provide a means
of comparison.

Masks are used to isolate specific features. A mask is a binary image, made by using
relational operators. A binary mask is multiplied by the original image to omit
specific areas. For more information, see “Masking Images’ on page 68.

Threshold levels can be applied to an image to clip the pixel valuesto afloor or a
ceiling. Clipping enhances specific features, and is applied through minimum and
maximum operators. After the resulting images are byte-scaled, the specific features
remain while the other areas become part of the background. For more information,
see “Clipping Images’ on page 72.

Locating pixel valuesis another way to segment specific features. Mathematical
expressions are used to determine the location of pixels with particular values within
the two-dimensional array representing the image. For more information, see
“Locating Pixel Valuesin an Image” on page 76.

When specific features have been segmented, image statistics (such as total, mean,
standard deviation, and variance) can be derived to quantify and compare them. For
more information, see “ Calculating Image Statistics’” on page 80.

Note
In this book, Direct Graphics examples are provided by default. Object Graphics
examples are provided in cases where significantly different methods are required.

The following list introduces image math operations and associated IDL math
operators and routines covered in this chapter.

Task Operator(s) and Routine(s) Description
“Masking Relational Operators Make masks and
Images” on Mathematical Operators apply them to
page 68. images.

Table 4-1: Image Math Tasks and Related Image Math Operators and
Routines

Task Operator(s) and Routine(s) Description
“Clipping Minimum and Maximum Operators | Clip the pixel values
Images’ on Mathematical Operators of an image to
page 72. highlight specific

features.

“Locating Pixel | WHERE L ocate specific
Valuesin an Mathematical Operators pixel values within
Image”’ on an image.
page 76.
“Calculating Mathematical Operators Calculate the sum,
Image IMAGE_STATISTICS mean, standard
Statistics” on deviation, and
page 80 variance of the pixel

values within an
image.

Table 4-1: Image Math Tasks and Related Image Math Operators and

Note

Routines (Continued)

67

This chapter uses datafilesfrom the IDL examples/data and
examples/demo/demodata directories. Twofiles, data.txt and index. txt,
contain descriptions of the files, including array sizes.

Masking Images

Masking (also known as thresholding) is used to isolate features within an image
above, below, or equal to a specified pixel value. The value (known as the threshold
level) determines how masking occurs. In IDL, masking is performed with the
relational operators. IDL’s relational operators are shown in the following table.

Operator Description
EQ Equal to
NE Not equal to
GE Greater than or equal to
GT Greater than
LE Lessthan or equal to
LT Lessthan

Table 4-2: IDL’s Relational Operators

For example, if you have an image variable and you want to mask it to include only
the pixel values equaling 125, the resulting mask variable is created with the
following IDL statement.

mask = image EQ 125

The mask level is applied to every element in the image array, which resultsin a
binary image.

Note
You can also provide both upper and lower bounds to masks by using the bitwise
operators; AND, NOT, OR, and XOR. See Bitwise Operatorsin the Building IDL
Applications for more information on these operators.

The following example uses masks derived from the image contained in the
worldelv.dat file, whichisinthe examples/data directory. Masks are derived
to extract the oceans and land. These masks are applied back to the image to show
only on the oceans or the land. Masks are applied by multiplying them with the
original image. Complete the following steps for a detailed description of the process.

Example Code
Seemaskingimages.pro inthe examples/doc/image subdirectory of the IDL

installation directory for code that duplicates this example.

1. Determine the path to thefile:

file = FILEPATH('worldelv.dat',6 $
SUBDIRECTORY = ['examples',6 'data'])

2. Initiaize the image size parameter:

imageSize = [360, 360]
3. Import the image from the file:

image = READ BINARY (file, DATA DIMS = imageSize)
4. Initialize the display:

DEVICE, DECOMPOSED = 0
LOADCT, 38

5. Create awindow and display the image:

WINDOW, 0, XSIZE = imageSize[0], YSIZE = imageSize[l], $
TITLE = 'World Elevation'
TV, image
The following figure shows the original image, which represents the elevation
levels of the world.

Figure 4-1: World Elevation Image

6. Make amask of the oceans:
oceanMask = image LT 125

7. Multiply the ocean mask by the original image:
maskedImage = image*oceanMask

8. Create another window and display the mask and the results of the
multiplication:

WINDOW, 1, XSIZE = 2*imageSize[0], YSIZE = imageSize[l], $
TITLE = 'Oceans Mask (left) and Resulting Image (right)'

TVSCL, oceanMask, 0

TV, maskedImage, 1

The following figure shows the mask of the world’'s oceans and the results of
applying it to the original image.

Figure 4-2: Oceans Mask (left) and the Resulting Image (right)

9. Makeamask of the land:
landMask = image GE 125
10. Multiply the land mask by the original image:

maskedImage = image*landMask

Masking Images Image Processing in IDL

11. Create another window and display the mask and the results of the
multiplication:

WINDOW, 2, XSIZE = 2*imageSize[0], YSIZE = imageSize[l], $
TITLE = 'Land Mask (left) and Resulting Image (right)'

TVSCL, landMask, O

TV, maskedImage, 1

The following figure shows the mask of the land masses of the world and the
results of applying it to the original image.

Figure 4-3: Land Mask (left) and the Resulting Image (right)

Clipping Images

Clipping is used to enhance features within an image. You provide athreshold level
to determine how the clipping occurs. The values above (or below) the threshold
level remain the same while the other values are set equal to the level.

In DL, clipping is performed with the minimum and maximum operators. IDL’s
minimum and maximum operators are shown in the following table.

Operator Description
< Less than or equal to
> Greater than or equal to

Table 4-3: IDL's Minimum and Maximum Operators

The operators are used in an expression that contains an image array, the operator,
and then the threshold level. For example, if you have an image variable and you
want to scale it to include only the values greater than or equal to 125, the resulting
clippedimage variable is created with the following IDL statement.

clippedImage = image > 125

The threshold level is applied to every element in the image array. If the element
valueislessthan 125, it is set equal to 125. If the value is greater than or equal to
125, it is left unchanged.

Note
When clipping is combined with byte-scaling, thisis equivalent to performing a
stretch on an image. See “Determining Intensity Values for Threshold and Stretch”
in Chapter 9 for more information.

Thefollowing example shows how to threshold an image of Hurricane Gilbert, which
isinthehurric.dat fileinthe examples/data directory. Two clipped images
are created. One contains all data values greater than 125 and the other contains all
values less than 125. Since these clipped images are grayscale images and do not use
the entire O to 255 range, they are displayed with the TV procedure and then scaled
with the TV SCL procedure, which scales the range of the image from 0 to 255.
Complete the following steps for a detailed description of the process.

Example Code
Seeclippingimages.pro inthe examples/doc/image subdirectory of the
IDL installation directory for code that duplicates this example.

1. Determinethe path to the worldtmp . png file:

file = FILEPATH('hurric.dat', $
SUBDIRECTORY = ['examples',6 'data'])

2. Definethe image size parameter:

imageSize = [440, 340]
3. Import the image from the file:

image = READ BINARY (file, DATA DIMS = imageSize)
4. Initialize the display:

DEVICE, DECOMPOSED = 0
LOADCT, O

5. Create awindow and display the image:

WINDOW, 0, XSIZE = imageSize[0], YSIZE = imageSize[l], $
TITLE = 'Hurricane Gilbert'
TV, image

The following figure shows the original image of Hurricane Gilbert.

Figure 4-4: Image of Hurricane Gilbert

6. Clip the image to determine which pixel values are greater than 125:

topClippedImage = image > 125

7. Create another window and display the clipped image with the TV (left) and
the TVSCL (right) procedures:

WINDOW, 1, XSIZE = 2*imageSize[0], YSIZE = imageSizel[l], $
TITLE = 'Image Greater Than 125, TV (left) ' + $
'and TVSCL (right)'

TV, topClippedImage, 0

TVSCL, topClippedImage, 1

The following figure shows the resulting image of pixel values greater than
125 with the TV and TV SCL procedures.

Figure 4-5: Pixel Values Greater Than 125, TV (left) and TVSCL (right)

8. Clip theimage to determine which pixel values are less than a 125:
bottomClippedImage = image < 125

9. Create another window and display the clipped image with the TV and the
TVSCL procedures:

WINDOW, 1, XSIZE = 2*imageSize[0], YSIZE = imageSizel[l], $
TITLE = 'Image Less Than 125, TV (left) ' + S
'and TVSCL (right)'

TV, bottomClippedImage, 0

TVSCL, bottomClippedImage, 1

The following figure shows the resulting image of pixel values lessthan 125
withthe TV (left) and TV SCL (right) procedures.

L - ﬁi“

Figure 4-6: Pixel Values Less Than 125, TV (left) and TVSCL (right)

Locating Pixel Values in an Image

L ocating pixel values within an image helps to segment features. You can use IDL's
WHERE function to determine where features characterized by specific values
appear within the image. The WHERE function returns a vector of one-dimensional
indices, locating where the specified values occur within the image. The values are
specified with an expression input argument to the WHERE function. The expression
is defined with the relational operators, similar to how masking is performed. See
“Masking Images’ on page 68 for more information on relational operators.

Since the WHERE function only returns the one-dimensional indices, you must
derive the column and row locations with the following statements.

column = index MOD imageSize[0]
row = index/imageSize[0]

where index is the result from the WHERE function and imageSze[ 0] is the width of
the image.

The WHERE function returns one-dimensional indicesto allow you to easily use
these results as subscripts within the original image array or another array. This
ability allows you to combine values from one image with another image. The
following example combines specific values from the image within the
worldelv.dat file with theimage within the worldtmp.png file. The
worldelv.dat fileisinthe examples/data directory and theworldtmp.png file
isinthe examples/demo/demodata directory. First, the temperature datais shown
in the oceans and the elevation data is shown on the land. Then, the elevation datais
shown in the oceans and the temperature data is shown on the land. Compl ete the
following steps for a detailed description of the process.

Example Code
See combiningimages.pro inthe examples/doc/image subdirectory of the
IDL installation directory for code that duplicates this example.

1. Determine the path to thefile:

file = FILEPATH('worldelv.dat',6 $
SUBDIRECTORY = ['examples', 'data'l)

2. Initialize theimage size parameter:
imageSize = [360, 360]
3. Import the elevation image from the file:

elvIimage = READ BINARY (file, DATA DIMS = imageSize)

4. Initialize the display:

DEVICE, DECOMPOSED = 0
LOADCT, 38

5. Create awindow and display the elevation image:

WINDOW, 0, XSIZE = 2*imageSize[0], YSIZE = imageSize[l], $
TITLE = 'World Elevation (left) and Temperature (right)'
TV, elvImage, O

6. Determine the path to the other file:

file = FILEPATH('worldtmp.png', $
SUBDIRECTORY = ['examples', 'demo', 'demodata'l])

7. Import the temperature image:
tmpImage = READ PNG(file)

8. Display the temperature image:
TV, tmpImage, 1

The following figure shows the origina world elevation and temperature
images.

Figure 4-7: World Elevation (left) and Temperature (right)

9. Determine where the oceans are located within the elevation image:

ocean = WHERE (elvImage LT 125)

10.

11.

12.

13.

14.

15.

16.

Set the temperature image as the background:
image = tmpImage

Replace values from the temperature image with the values from the elevation
image only where the ocean pixels are located:

image [ocean] = elvImage [ocean]
Create another window and display the resulting temperature over land image:

WINDOW, 1, XSIZE = 2*imageSize[0], YSIZE = imageSize[l], $
TITLE = 'Temperature Over Land (left) ' +
'and Over Oceans (right)'

TV, image, O

Determine where the land is located within the el evation image:
land = WHERE (elvImage GE 125)

Set the temperature image as the background:
image = tmpImage

Replace values from the temperature image with the values from the elevation
image only where the land pixels are located:

image [land] = elvImage [land]
Display the resulting temperature over oceans image:

TV, image, 1

The following figure shows two possible image combinations using the world
elevation and temperature images.

Figure 4-8: Temperature Over Land (left) and Over Oceans (right)

Tip
You could also construct the same image using masks and adding them together.
For example, to create the second image (temperature over oceans), you could have
done the following:

mask = elvImage GE 125
image = (tmpImage* (1 - mask)) + (elvImage*mask)

For large images, using masks may be faster than using the WHERE routine.

Calculating Image Statistics

The statistical properties of an image provide useful information, such as the total,
mean, standard deviation, and variance of the pixel values. IDL’s
IMAGE_STATISTICS procedure can be used to calculate these statistical properties.
The MOMENT, N_ELEMENTS, TOTAL, MAX, MEAN, MIN, STDDEV, and
VARIANCE routines can also be used to calculate individual statistics, but most of
these values are already provided by the IMAGE_STATISTICS procedure.

The following example shows how to use the IMAGE_STATISTICS procedure to
calculate the statistical properties of an image. First, amask is used to subtract the
convection of the earth’s core from the convection image contained in the
convec.dat file, whichisin the examples/data directory. Theresulting
difference represents the convection of just the earth’s mantle. The
IMAGE_STATISTICS procedure is applied to this difference image, and the
resulting values are displayed in the Output Log. Then, amask is derived for the non-
zero values of the difference image, and the IMAGE_STATISTICS procedure is used
again, this time with the mask applied through the MASK keyword. The resulting
statistics can than be compared. The color table associated with this example iswhite
for zero values and dark red for 255 values. Complete the following steps for a
detailed description of the process.

Example Code
Seecalculatingstatistics.prointhe examples/doc/image subdirectory
of the IDL installation directory for code that duplicates this example.

1. Determine the path to thefile:

file = FILEPATH('convec.dat',6 $
SUBDIRECTORY = ['examples',6 'data'])

2. Initialize the image size parameter.

imageSize = [248, 248]
3. Import the image from thefile:

image = READ BINARY (file, DATA DIMS = imageSize)
4. Initialize the display:

DEVICE, DECOMPOSED = 0
LOADCT, 27

5. Create awindow and display the image:

WINDOW, 0, XSIZE = imageSize[0], YSIZE = imageSize[l], $
TITLE = 'Earth Mantle Convection'
TV, image

The following figure shows the original convection image.

Figure 4-9: Earth Mantle Convection

6. Makeamask of the core and scaleit to range from O to 255:
core = BYTSCL (image EQ 255)

7. Subtract the scaled mask from the original image:

difference = image - core
8. Create another window and display the difference of the original image and the
scaled mask:
WINDOW, 2, XSIZE = imageSize[0], YSIZE = imageSize[l], $
TITLE = 'Difference of Original & Core'

TV, difference

The following figure shows the convection of just the earth’s mantle.

Figure 4-10: The Difference of the Original Image and the Core

9. Determine the statistics of the difference image:

IMAGE _STATISTICS, difference, COUNT = pixelNumber, $
DATA SUM = pixelTotal, MAXIMUM = pixelMax, $
MEAN = pixelMean, MINIMUM = pixelMin, $
STDDEV = pixelDeviation, $
SUM_OF SQUARES = pixelSquareSum, $
VARIANCE = pixelVariance

10. Print out the resulting statistics:

PRINT, ''

PRINT, 'IMAGE STATISTICS:'

PRINT, 'Total Number of Pixels = ', pixelNumber

PRINT, 'Total of Pixel Values = ', pixelTotal

PRINT, 'Maximum Pixel Value = ', pixelMax

PRINT, 'Mean of Pixel Values = ', pixelMean

PRINT, 'Minimum Pixel Value = ', pixelMin

PRINT, 'Standard Deviation of Pixel Values = ', $
pixelDeviation

PRINT, 'Total of Squared Pixel Values = ', $
pixelSquareSum

PRINT, 'Variance of Pixel Values = ', pixelVariance

IDL prints:

IMAGE STATISTICS:

Total Number of Pixels = 61504

Total of Pixel Values = 2.61691e+006

Maximum Pixel Value = 253.000

Mean of Pixel Values = 42.5486

Minimum Pixel Value = 0.000000

Calculating Image Statistics

Standard Deviation of Pixel Values = 48.7946
Total of Squared Pixel Values = 2.57779e+008
Variance of Pixel Values = 2380.91

11. Derive amask of the non-zero values of the image:
nonzeroMask = difference NE 0

12. Determine the statistics of the image with the mask applied:

IMAGE STATISTICS, difference, COUNT = pixelNumber, $
DATA SUM = pixelTotal, MASK = nonzeroMask, $
MAXIMUM = pixelMax, MEAN = pixelMean, $
MINIMUM = pixelMin, STDDEV = pixelDeviation, $
SUM_OF SQUARES = pixelSquareSum, $
VARIANCE = pixelVariance

13. Print out the resulting statistics:

PRINT, "'

PRINT, 'MASKED IMAGE STATISTICS:'

PRINT, 'Total Number of Pixels = ', pixelNumber

PRINT, 'Total of Pixel Values = ', pixelTotal

PRINT, 'Maximum Pixel Value = ', pixelMax

PRINT, 'Mean of Pixel Values = ', pixelMean

PRINT, 'Minimum Pixel Value = ', pixelMin

PRINT, 'Standard Deviation of Pixel Values = ', $
pixelDeviation

PRINT, 'Total of Squared Pixel Values = ', $
pixelSquareSum

PRINT, 'Variance of Pixel Values = ', pixelVariance

IDL prints:

MASKED IMAGE STATISTICS:

Total Number of Pixels = 36325

Total of Pixel Values = 2.61691e+006

Maximum Pixel Value = 253.000

Mean of Pixel Values = 72.0416

Minimum Pixel Value = 1.00000

Standard Deviation of Pixel Values = 43.6638
Total of Squared Pixel Values = 2.57779e+008
Variance of Pixel Values = 1906.53

The difference in the resulting statistics are because of the zero values, which
are apart of the calculations for the image before the mask is applied.

Chapter 5
Warping Images

This chapter describes the following topics:

Overview of WarpingImages ........... 86 Warping Images Using Direct Graphics... 89
Creating Transparent Image Overlays . . . .. 88

Overview of Warping Images

In image processing, image warping is used primarily to correct optical distortions
introduced by camera lenses, or to register images acquired from either different
perspectives or different sensors. When correcting optical distortions, the original
image may be registered to aregular grid rather than to another image. In image
warping, corresponding control points (selected in the input and reference images)
control the geometry of the warping transformation. The arrays of control pointsfrom
the original input image, Xi and i, are stretched to conform to the control point
arrays Xo and Yo, designated in the reference image. Because these transformations
are frequently nonlinear, image warping is often known as rubber sheeting. For
general tips regarding control point selection see “ Tips for Selecting Control Points”
on page 87.

Image warping in IDL is athree-step process. First, control points are selected
between two displayed images or between an image and agrid. Second, the resulting
arrays of control points, Xi, Yi, Xo, and Yo, are then input into one of IDL’s warping
routines. Third, the warped image resulting from the trandlation of the Xi, Yi pointsto
the Xo, Yo points, is displayed. It is often useful to display the warped image as a
transparency, overlaying the reference image. For more information on creating
transparencies with Direct and Object Graphics, see “ Creating Transparent Image
Overlays’ on page 88.

The following table introduces the tasks and routines covered in this chapter.

Task Routine Description
Creating aDirect | WSET Set the window focus and select control
GraphicsDisplay | cURSOR point coordinates.
of Image Warping _ _
eorwapng | VARPTR! - ap el WARP TR
Images Using 9 P '
Direct Graphics” | POLYWARP Create arrays of polynomial coefficients
on page 89. from the control point arrays before
using POLY _2D.
POLY_2D Warp the images using the polynomial
warping functions of POLY_2D.
XPALETTE Use XPALETTE to view acolor table.

Table 5-1: Image Warping Tasks and Routines

87
Task Routine Description
Creating an IDLgrPalette::Init Create a pal ette object.
Object Graphics ; ;
Display of Image XROI StelI ict control points using the XROI
Warping utility.
See “Warping WARP_TRI Warp the input image to the reference
Image Objects’ in image using the triangulation and
Chapter 4 of the interpolation functions of WARP_TRI.
Object SIZE Change the warped image into a RGB
Programming BYTARR image containing an alpha channel to
manual. enable transparency.
IDLgrImage:Init Initialize transparent image and base
image objects.
IDLgrWindow::Init | Initialize the objects necessary for an
IDLgrview:Init | Object Graphics display.
IDLgrModel::Init
Table 5-1: Image Warping Tasks and Routines (Continued)

Tips for Selecting Control Points

Both examplesin this chapter use control points to define the image warping
transformation. To produce accurate results, use the following guidelines when
selecting corresponding control points:

e Select numerous control points. A warping transformation based on many
control points produces a more accurate result than one based on only afew

control paints.

*  Select control points near the edges of the image in addition to control points
near the center of the image.

e Select ahigher density of control pointsin irregular or highly varying areas of

the image.

e Select pointsin which you are confident. Including points with poor accuracy
may generate worse results then awarp model with fewer points.

Image Processing in IDL

Creating Transparent Image Overlays

It is possible to create and display atransparent image using either IDL Direct
Graphics or IDL Object Graphics. Creating a transparent image is useful in the
warping process when you want to overlay atransparency of the warped image onto
the reference image (the image in which Xo, Yo control points were selected). The
method used to create and display the transparent image depends on whether the
resulting image is being displayed with Direct Graphics or Object Graphics.

Displaying Image Transparencies Using Direct
Graphics

Creating a transparent overlay in Direct Graphics requires devising a mask to alter
the array of the image that isto be displayed as atransparency. The mask retains only
the pixel values that will appear in the transparent overlay. The base image and the
transparent warped image can then be displayed as a blended image in a Direct
Graphics window.

With Direct Graphics displays, only asingle color table can be applied to the blended
image in adisplay window. For an example of creating a blended image, combining a
warped image and a base image, see “Warping Images Using Direct Graphics’ on
page 89.

Note
For precise control over the color tables associated with the reference image and the
warped image transparency, consider using Object Graphics.

Displaying Image Transparencies Using Object
Graphics
In Object Graphics, atransparent image object is created by adding an alpha channel
to the image array. The alpha channel is used to define the level of transparency in an

image object. For an example, see “Defining Transparency in Image Objects’ and
“Warping Image Objects’ in Chapter 4 of the Object Programming manual.

Creating Transparent Image Overlays Image Processing in IDL

Warping Images Using Direct Graphics

Image warping requires selection of corresponding control pointsin an input image
and either areference image or aregular grid. The input image is warped so that the
input image control points match the control points specified in the reference image.

Using Direct Graphics, the following example warps the input image, a Magnetic
Resonance Image (MRI) proton density scan of a human thoracic cavity, to the
reference image, a Computed Tomography (CT) bone scan of the same region.
Complete the following steps for a detailed description of the process.

Example Code
Seemriwarping direct.prointheexamples/doc/image subdirectory of the
IDL installation directory for code that duplicates this example.

1. Select the MRI proton density image file:

mriFile= FILEPATH('pdthoraxl24.jpg', $
SUBDIRECTORY = ['examples',6 'data'])

2. Use READ_JPEG to read in the input image, which will be warped to the CT
bone scan image. Then prepare the display device, load agrayscale color table,
create awindow and display the image:

READ JPEG, mriFile, mriImg
DEVICE, DECOMPOSED = 0

LOADCT, O
WINDOW, 0, XSIZE = 256, YSIZE = 256, $
TITLE = 'MRI Proton Density Input Image'

TV, mriImg
3. Select the CT bone scan imagefile:

ctboneFile = FILEPATH('ctbonel57.jpg', $
SUBDIRECTORY = ['examples',6 'data'])

4. Use READ_JPEG to read in the reference image and create awindow:

READ JPEG, ctboneFile, ctbonelImg
WINDOW, 2, XSIZE = 483, YSIZE = 410, $
TITLE = 'CT Bone Scan Reference Image'

5. Loadthe“Hue Sat Lightness 2" color table, making theimage's features easier
to distinguish. After displaying the image, return to the gray scale color table.

LOADCT, 20
TV, ctbonelImg
LOADCT, O

Image Processing in IDL Warping Images Using Direct Graphics

Proceed with the following section to begin selecting control points.
Direct Graphics Example: Selecting Control Points

This section describes selecting corresponding control pointsin the two displayed
images. The array of control points (Xi, Yi) in the input image will be mapped to the
array of points (Xo, Yo) selected in the reference image. The following image shows
the points to be selected in the input image.

CP4
(xi4, yi4)
CP3 CP5
(xi3, yi3) (xi5, yi5)
CP2 CP6
(xi2, yi2) —— (xi6, yi6)
CP1
xil, yil CP7
(L, yid) (xi7, yi7)
CP9 CP 8
(xi9, yi9) (xi8, yi8)

Figure 5-1: Control Points (CP) Selection in the Input Image

1. Set focus on the first image window:
WSET, O

2. Select thefirst control point using the CURSOR function. After entering the
following line, the cursor changes to a cross hair when positioned over the
image window. Position the cross hair so that it is on thefirst control point,
“CP 1", depicted by awhite circle in the lower-left corned of the previous

Warping Images Using Direct Graphics Image Processing in IDL

figure, and click the left mouse button. The x, y coordinate values of the first
control point will be saved in the variables xi1, yil:

CURSOR, xil, yil, /DEVICE

Note
Thevaluesfor xil and yil are displayed in the IDLDE Variable Watch
window. If you are not running the IDLDE, you can type PRINT, xil,
yvil toseethevalues.

Note
After entering the first line and selecting the first control point in the display
window, place your cursor in the IDL command line and press the Up Arrow
key. Thelast line entered is displayed and can be easily modified.

3. Continue selecting control points. After you enter each of the following lines,
select the appropriate control point in the input image as shown in the previous
figure:

CURSOR, xi2, yi2, /DEVICE
CURSOR, xi3, yi3, /DEVICE
CURSOR, xi4, yi4, /DEVICE
CURSOR, xi5, yi5, /DEVICE
CURSOR, xi6, yi6, /DEVICE
CURSOR, xi7, yi7, /DEVICE
CURSOR, xi8, yi8, /DEVICE
CURSOR, xi9, yi9, /DEVICE

4. Set the focus on the window containing the reference image to prepare to
select corresponding control points:

WSET, 2

Note
The Xi and ¥Yi vectors and the Xo and Yo vectors must be the same length,
meaning that you must select the same number of control pointsin the
reference image as you selected in the input image. The control points must
also be selected in the same order since the point Xi1, Yil will be warped to
Xol, Yol.

The following figure displays the control pointsto be selected in the next step.

CP4
(x4, yad)
CP 3 CP5
(x03, yo3) & LR (x05, yo5)
CP2 : ) - CP#6
(x02, yo2} p o & - (%06, yoB)
CP1 CP7
xol, yol) — p g & 4— (X07, yo7)
CP9 CPs8
(x09, yo9) (x08, yo8)

Figure 5-2: Control Point (CP) Selection in the Reference Image

5. Select the control pointsin the reference image. These are the corresponding
points to which the input image control pointswill be warped. After entering
each line, select the appropriate control point as shown in the previous figure;

CURSOR, xol, yol,
CURSOR, x02, yo2,
CURSOR, x03, yo3,
CURSOR, x04, yo4,
CURSOR, x05, yo5,
CURSOR, x06, yo6,
CURSOR, xo07, yo7,
CURSOR, x08, yo8,
CURSOR, x09, yo9,

/DEVICE
/DEVICE
/DEVICE
/DEVICE
/DEVICE
/DEVICE
/DEVICE
/DEVICE
/DEVICE

6. Placethe control pointsinto vectors (one-dimensional arrays) required by IDL
warping routines. WARP_TRI and POLY WARP use the variables Xi, Yi and
Xo, Yo as containers for the control points selected in the original input and
reference images. Geometric transformations control the warping of the input

Warping Images Using Direct Graphics

image (Xi, Yi) valuesto the reference image (Xo, Yo) values. Enter the
following lines to load the control point values into the one-dimensional

arrays:
Xi = [x1il1, xi2, xi3, xi4, x1i5, xi6, xi7, xi8, x19]
Yi = [yil, yi2, yi3, yi4, yi5, yie6, yi7, yi8, yio9l
Xo = [x0l, x02, x03, x04, x05, xX06, x07, X08, x09]
Yo = [yol, yo2, yo3, yo4, yo5, yo6, yo7, yo8, yo9l

Example Code: Warping and Displaying a Transparent Image
Using Direct Graphics

This section uses the control points defined in the previous section to warp the
original MRI scan to the CT scan, using both of IDL’s warping routines, WARP_TRI
and POLY _2D. After outputting the warped image, it will be atered for display asa
transparency in Direct Graphics.

1. Warp the input image, mrilmg, onto the reference image using WARP_TRI.
Thisfunction usestheirregular grid of the reference image, defined by Xo, Yo,
as abasis for triangulation, defining the surfaces associated with (Xo, Yo, Xi)
and (Xo, Yo, Y). Each pixel in the input image is then transferred to the
appropriate position in the resulting output image as designated by
interpolation. Using the WARP_TRI syntax,

Result = WARP_TRI(Xo, Yo, Xi, Yi, Image, OUTPUT SIZE=vector]
[, /QUINTIC] [, /EXTRAPOLATE])

set the OUTPUT _SIZE equal to the reference image dimensions since this
image forms the basis of the warped, output image. Use the EXTRAPOLATE
keyword to display the portions of the image which fall outside of the
boundary of the selected control points:

warpTriImg = WARP_TRI (Xo, Yo, Xi, Yi, mriImg, $
OUTPUT_SIZE=[483, 410], /EXTRAPOLATE)

Note
Images requiring more aggressive warp models may not have good results
outside of the extent of the control points when WARP_TRI is used with the
/IEXTRAPOLATE keyword.

2. Create anew window and display the warped image:

WINDOW, 3, XSIZE = 483, YSIZE = 410, TITLE = 'WARP TRI image'
TV, warpTriImg

Image Processing in IDL Warping Images Using Direct Graphics

You can see the how precisely the control points were selected by the amount
of distortion in the resulting warped image. The following figure showslittle
distortion.

1 WARP_TRI image
—

Figure 5-3: Warped Image Produced with WARP_TRI

3. Use POLYWARP in conjunction with POLY _2D to create another warped
image for comparison with the WARP_TRI image. First use the POLY WARP
procedure to create arrays (p, g) containing the polynomial coefficients
required by the POLY _2D function:

POLYWARP, Xi, Yi, Xo, Yo, 1, p, g

4. Usingthep, q array values generated by POLY WARP, warp the original
image, mrilmg, onto the CT bone scan using the POLY _2D function syntax,

Result = POLY 2D( Array, P, Q [, Interp [, Dimx, Dimy]]
[, CUBIC={-1 to 0}] [, MISSING=value] )

Specify avalue of 1 for the Interp argument to use bilinear interpolation and
set pimx, Dimy equal to thereference image dimensions:

warpPolyImg = POLY 2D(mriImg, p, g, 1, 483, 410)
5. Create anew window and display the image created using POLY _2D:

WINDOW, 4, XSIZE = 483, YSIZE = 410, TITLE = 'Poly 2D image'
TV, warpPolyImg

The following image shows little difference from the WARP_TRI image other
than more accurate placement in the display window.

! Poly_2D image

Figure 5-4: Warped Image Produced with POLY_2D

Direct Graphicsdisplaysin IDL allow you to display a combination of images
in the same Direct Graphics window. The following steps display various
intensities of the warped image and the reference image in a Direct Graphics
window.

6. Usethe XPALETTE tool to view the color table applied to the bone scan
image by first entering:

XPALETTE

In the XPALETTE utility, display acolor table by selecting the Predefined
button. In the resulting XLOADCT dialog, scroll down and select Hue
Saturation Lightness 2. Click Done. Inthe XPALETTE utility, click
Redraw. Compare the bone scan image, displayed in window 2, to the
displayed color table. To mask out the less important background information,
select acolor close to that of the body color in the image.

Image Processing in IDL Warping Images Using Direct Graphics

The following figure displays a portion of the XPALETTE utility with such a
selection.

55
. | I
By Index
Fow
7
: | I
Tolurn

Figure 5-5: Using XPALETTE to ldentify Mask Values

7. Using the knowledge that the body color’'sindex number is 55, mask out the
lessimportant background information of the bone scan image by creating an
array containing only pixel values greater than 55. Multiply the mask by the
image to retain the color information and use BY TSCL to scale the resulting
array from 0 to 255:

ctboneMask = BYTSCL( (ctboneImg GT 55) * ctbonelImg)

8. Display ablended image using the full intensity of the bone scan image and a
75% intensity of the warped image. The following statement displays the
pixelsin the bone scan with the full range of colorsin the color table while
using the lower 75% of the color table values for the warped image. After
adding the arrays, scale the results for display purposes:

blendImg = BYTSCL (ctboneMask + 0.75 * warpPolyImg)
9. Create awindow and display the result:

WINDOW, 5, XSIZE = 483, YSIZE = 410, TITLE = 'Blended Image'
TV, blendImg

Warping Images Using Direct Graphics Image Processing in IDL

The clavicles and rib bones of the reference image are clearly displayed in the
following figure.

Figure 5-6: Direct Graphics Display of a Transparent Blended Image

While Direct Graphics supports displaying indexed images as transparent blended
images, you could also apply apha blending to RGB images that are output to a
TrueColor display. However, creating image transparencies which retain their color
information is more easily accomplished using Object Graphics. For an example of
using Object Graphicsto display awarped image transparency over another image
see “Warping Image Objects’ in Chapter 4 of the Object Programming manual.

Chapter 6

Working with Regions
of Interest (ROISs)

This chapter describes creating and analyzing regions of interest (ROIs) and includesthe following

topics:

Overview of WorkingwithROIs ... ..... 100 GrowingaRegion .................. 113
Defining Regions of Interest ........... 103 Creating and Displaying an ROl Mask .. 118
Displaying ROI Objectsin aDirect Graphics Testing an ROI for Point Containment ... 122
Window .............. .. i 105  Cresating a Surface Mesh of an ROl Group 125

Programmatically Defining ROIs

Overview of Working with ROls

A region of interest (ROI) isan area of an image defined for further analysis or
processing. There are several ways to define ROIs. The XROI utility enables the
interactive definition of single or multiple regions from an image using the mouse.
Routines such as CONTOUR or REGION_GROW enabl e the programmatic
definition of ROIS. CONTOUR traces the outlines of thresholded ROIs while the
REGION_GROW routine expands an initial region to include all connected,
neighboring pixels that meet given conditions. Once an ROI is defined, it can be
displayed or undergo further analysis.

An ROI can be displayed using either Direct Graphics or Object Graphics. In Direct
Graphics, the DRAW_ROI routine quickly displays single or multiple ROI objects or
an ROI group. In Object Graphics, see IDLanROI and IDLgrROI in the IDL
Reference Guide for more information.

Note
When computing ROI geometry, there is a difference between aregion’s areawhen
it isdisplayed on a screen versus the region’s computed, geometric area. See
“Contrasting an ROI's Geometric Areaand Mask Area” on page 101 for details.

Multiple ROIs can also be defined from a multi-image data set and added to an
IDLanROIGroup object for triangulation into a 3D mesh. Alternatively, multiple
ROl s can be defined in a single image and added to a group object. ROl groups can
be displayed in a Direct Graphics window with DRAW_ROI or with the Object
Graphics XOBJVIEW tility.

The following table introduces the tasks and routines covered in this chapter.

Task Routine(s)/Object(s) Description
“Defining Regions | XROI Create an ROI
of Interest” on interactively, prior to
page 103. analysis or display.
“Displaying ROI DRAW_ROI Display ROI objectsin a
Objectsin aDirect Direct Graphics window.
Graphics Window”
on page 105.

Table 6-1: Tasks and Routines Associated with Regions of Interest

Overview of Working with ROIs Image Processing in IDL

Task

Routine(s)/Object(s)

Description

“Programmatically
Defining ROIS’ on

CONTOUR
DRAW_ROI

Define ROIs using
CONTOUR and display

page 109. IDLanROI::ComputeM ask them using DRAW_ROI.
IMAGE_STATISTICS Return various statistics
IDLanROI::ComputeGeometry for each ROI.

“Growing a REGION_GROW Expand an original region
Region” on to include al connected,
page 113. neighboring pixels which

meet specified
constraints.
“Creating and IDLanROI::ComputeMask Create a2D mask of an
Displaying an ROI ROI, compute the area of

Mask” on page 118.

the mask and display a
magnified view of the
image region.

“Testing an ROI for
Point Containment”
on page 122.

IDLanROI::ContainsPoints

Determine whether a
point lies within the
boundary of aregion.

“Creating a Surface
Mesh of an ROI
Group” on

page 125.

IDLanROIGroup::Add
IDLanROIGroup::ComputeM esh
XOBWVIEW

Add ROIsto an ROI
group object, triangulate
a surface mesh and
display the group object
using XOBJVIEW.

Table 6-1: Tasks and Routines Associated with Regions of Interest (Continued)

Contrasting an ROI's Geometric Area and Mask Area

When working with ROIs, many users note a discrepancy between the computation
of an ROI’s geometric area and the computation of the mask area (the number of
pixels an ROI contains when displayed). Intuition might lead one to believe that the
results should be the same. However, as the following figure shows, the computed
geometric area (the result of a pure mathematical calculation) differs from the
displayed (masked) area, which is subject to the artifacts of digital sampling.

When displaying aregion (or computing the area of its mask), each vertex of the
region is mapped to a corresponding discrete pixel location. No matter where the

Image Processing in IDL

vertex falls within the pixel, the entire pixel location is set since the region is being
displayed. For example, for any vertex coordinate (X, y) where:

1.5 <x <2.5and 1.5 <y < 2.5

the vertex coordinate is assigned avalue of (2, 2). Therefore, the area of the displayed
(masked) region istypically larger than the computed geometric area. While the
geometric area of a2 by 2 region equals 4 as expected, the mask area of the identical
region equals 9 due to the centering of the pixels when the region is displayed.

‘{2-4) (4,4) 2-43 2-4)
4%
yd o4
/ //
(22) T (.2.2 (1.2}

Geometric Area
2 x 2region =4 e | Screen Display of Same Region
.{0.0) (0,0 3 x 3 filled region =9

Figure 6-1: A Region’s Undisplayed Area (left) vs. Displayed Area (right)

The ROI Information dialog of the XROI utility reports the region’s “Area’
(geometric area) and “# Pixels’ (mask area). To programmatically compute an ROI’s
geometric area, use IDLanROI::ComputeGeometry. To programmatically compute
the area of adisplayed region, use IDLanROI::ComputeMask in conjunction with
IMAGE_STATISTICS. See “Programmatically Defining ROIS” on page 109 for
examples of these computations.

Defining Regions of Interest

The XROI utility allows you to quickly load an image file, define single or multiple
ROIs, and obtain geometry and statistical data about the ROIs. While regions can be
defined programmatically (see “ Programmatically Defining ROIS’ on page 109 and

“Growing aRegion” on page 113), the XROI utility enables the interactive creation

and selection of an ROI using the mouse.

For a quick introduction to creating ROIs using XROI, complete the following steps:
1. Open XROI by typing the following at the command line:
XROI

2. Load animage using the image file selection dialog. Select earth. jpg from
the examples/demo/demodata directory. Click Open.Theimage appearsin
the XROI utility.

See “Using XROI” under “XROI"” in the IDL Reference Guide manual for
details on the interface elements.Flip the image vertically to display it right-
side-up by clicking the Flip button.

3. Select the Draw Freehand button and use the mouse to interactively define an
ROI encompassing the African continent. Your image should be similar to the
following figure.

&1 ROI o [=] 3
File Edit

‘Elﬂ Byl x|o|o| v | @l ROI Information H=

Fiegions of Interest:

=t Arear 28195000

Perimeter:  238.21336
# Pirels: 2967
Minimum: — MNAA
M awimum: A&
Mean: WA
Std. Dev.: MNAA

B Mame:|Fegion 1 Delete ROI
Cloze | Histogram |

Figure 6-2: Defining an ROI of Africa and Showing the ROI Information Dialog

Image Processing in IDL Defining Regions of Interest

4. After releasing the mouse button, the ROI Information dialog appears,
displaying ROI statistics. You can nhow define another ROI, save the defined
ROI asa . sav file or exit the XROI utility.

Using XROI syntax alows you to programmatically load an image and specify a
variablefor REGIONS_OUT that will contain the ROI data. The region data can then
undergo further analysis and processing. The following code lines open the
previously opened image for ROI creation and selection and specify to save the
region data as oROI Africa.

; Select the file, read the data and load the image’s color table.
imgFile = FILEPATH('earth.jpg', $
SUBDIRECTORY = ['examples', 'demo', 'demodata'l])
image = READ IMAGE (imgFile, R, G, B)
TVLCT, R, G, B

; Display the image using XROI. Specify a variable for REGIONS_ OUT
; to save the ROI information.
XROI, image, R, G, B, REGIONS OUT = oROIAfrica

The ROI information, oROI Africa, can then be analyzed using | DLanROI methods or
the REGION_GROW procedure. The ROI data can aso be displayed using

DRAW_ROI or asan IDLgrROI abject. Such tasks are covered in the following
sections.

Defining Regions of Interest Image Processing in IDL

Displaying ROI Objects in a Direct Graphics
Window

The DRAW_ROI procedure displays single or multiple IDLanROI objectsin aDirect
Graphics window. The procedure allows you to layer the ROIs over the original
image and specify the line style and color with which each region is drawn. The
DRAW_ROI procedure also provides a means of easily displaying interior regions or
“holes” within adefined ROI.

The following example uses the XROI utility to define two regions, afemur and tibia
from a DICOM image of aknee, and draws them in a Direct Graphics window.
Complete the following steps for a detailed description of the process.

Example Code
See drawroiex.pro inthe examples/doc/image subdirectory of the IDL
installation directory for code that duplicates this example.

1. Preparethe display device and load a grayscale color table:

DEVICE, DECOMPOSED = 0, RETAIN = 2
LOADCT, O

2. Select and open the image file using the READ_DICOM function and get its
size:

kneeImg = READ DICOM(FILEPATH('mr knee.dcm', $
SUBDIRECTORY = ['examples',6 'data']))
dims = SIZE(kneeImg, /DIMENSIONS)

3. Rotate theimage 180 degrees so that the femur will be at the top of the display:
kneeImg = ROTATE (BYTSCL (kneelImg), 2)

4. Open thefilein the XROI utility to create an ROI containing the femur. The
following line includes the ROI_GEOMETRY and STATISTICS keywords so
that specific ROI information can be retained for printing in alater step:

XROI, kneeImg, REGIONS OUT = femurROIout, $
ROI_GEOMETRY = femurGeom, $
STATISTICS = femurStats, /BLOCK

Select the Draw Polygon button from the XROI utility toolbar, shown in the
following figure. Position the crosshairs anywhere along the border of the
femur and click the left mouse button to begin defining the ROI. Move your
mouse to another point along the border and left-click again. Repeat the
process until you have defined the outline for the ROI. To close the region,

Image Processing in IDL Displaying ROI Objects in a Direct Graphics Window

double-click the left mouse button. Your display should appear similar to the
following figure. Close the XROI utility to store the ROI information in the
variable, femurROlout.

ERERREERER T Draw Polygon

Figure 6-3: Defining the Femur ROI

5. Create an ROI containing the tibia, using the following XROI statement:

XROI, kneeImg, REGIONS OUT = tibiaROIout, $
ROI_GEOMETRY = tibiaGeom, $
STATISTICS = tibiaStats, /BLOCK

Select the Draw Polygon button from the XROI utility toolbar. Position the
crosshairs symbol anywhere along the border of the tibia and draw the region
shown in the following figure, repeating the same steps as those used to define
the femur ROI. Close the XROI utility to store the ROl information in the
specified variables.

Figure 6-4: Defining the Tibia ROI

6. Create aDirect Graphics display containing the original image:

WINDOW, 0, XSIZE = dims[0], YSIZE = dims[1]
TVSCL, kneelImg

7. Loadthe 16-level color table to display the regions using different colors. Use
DRAW_ROI statements to specify how each ROI is drawn:

LOADCT, 12

DRAW ROI, femurROIout, /LINE FILL, COLOR = 80, $
SPACING = 0.1, ORIENTATION = 315, /DEVICE

DRAW ROI, tibiaROIout, /LINE FILL, COLOR = 42, $
SPACING = 0.1, ORIENTATION = 30, /DEVICE

In the previous statements, the ORIENTATION keyword specifies the degree
of rotation of the lines used to fill the drawn regions. The DEVICE keyword
indicates that the vertices of the regions are defined in terms of the device
coordinate system where the origin (0,0) isin the lower-left corner of the
display.

Your results should appear similar to the following figure, with the ROI
objects layered over the original image.

Figure 6-5: Defined Region Objects Overlaid onto Original Image

8. Print the statistics for the femur and tibia ROIs. This information has been
stored in the femur Geom, femur Sat, tibiaGeom and tibiaStat variable
structures, defined in the previous XROI statements. Use the following linesto

print geometrical and statistical datafor each ROI:

PRINT, 'FEMUR Region Geometry and Statistics'

PRINT, 'area =', femurGeom.area, $
'perimeter = ', femurGeom.perimeter,
'population =', femurStats.count

PRINT, ' !

PRINT, 'TIBIA Region Geometry and Statistics'

PRINT, 'area =', tibiaGeom.area, $
'perimeter = ', tibiaGeom.perimeter,
'population =', tibiaStats.count

Note

Notice the difference between the “area” value, indicating the region’s
geometric area, and the “population” value, indicating the number of pixels
covered by the region when it is displayed. Thisdifferenceis expected andis
explained in the section, “Contrasting an ROI’'s Geometric Area and Mask

Ared’ on page 101.

9. Clean up object references that are not destroyed by the window manager

when you close the Object Graphics displays:

OBJ_DESTROY, [femurROIout, tibiaROIout]

Displaying ROI Objects in a Direct Graphics Window

Programmatically Defining ROIs

While most examplesin this chapter use interactive methods to define ROl s, aregion
can also be defined programmatically. The following example uses thresholding and
the CONTOUR function to programmatically trace region outlines. After the path
information of the regions has been input into ROI objects, the DRAW_ROI
procedure displays each region. The example then computes and returns the
geometric area and perimeter of each region as well as the number of pixels making
up each region when it is displayed. Complete the following steps for a detailed
description of the process.

Example Code
Seeprogramdefineroi.pro inthe examples/doc/image subdirectory of the

IDL installation directory for code that duplicates this example.

1. Preparethedisplay device and load a color table:

DEVICE, DECOMPOSED = 0, RETAIN = 2
LOADCT, O

2. Select and open the image file and get its dimensions:

img = READ PNG(FILEPATH('mineral.png', $
SUBDIRECTORY = ['examples', 'data'l))
dims = SIZE(img, /DIMENSIONS)

3. Create awindow and display the original image:

WINDOW, 0, XSIZE = dims[0], YSIZE = dims[1]
TVSCL, img

The following figure displays the initial image.

Figure 6-6: Initial Image

Create amask that identifies the darkest pixels, whose values are | ess than 50:

threshImg = (img LT 50)

Note

See “Determining Intensity Values for Threshold and Stretch” on page 243
for auseful strategy to use when determining threshold values.

Create and apply a 3x3 sguare structuring element, using the erosion and
dilation operators to close gaps in the thresholded image:

strucElem REPLICATE (1, 3, 3)

threshImg = ERODE (DILATE (TEMPORARY (threshImg), $
strucElem), strucElem)

Use the CONTOUR procedure to extract the boundaries of the threshol ded
regions. Store the path information and coordinates of the contours in the
variables pathlnfo and pathXY as follows:

CONTOUR, threshImg, LEVEL = 1, $
XMARGIN = [0, 0], YMARGIN = [0, 0], $
/NOERASE, PATH INFO = pathInfo, PATH XY = pathXY, $
XSTYLE = 5, YSTYLE = 5, /PATH DATA COORDS

The PATH_INFO variable contains the path information for the contours.
When used in conjunction with the PATH_XY variable, containing the
coordinates of the contours, the CONTOUR procedure records the outline of
closed regions. See CONTOUR in the IDL Reference Guide for full details.

Display the original image in a second window and load a discrete color table:

WINDOW, 2, XSIZE = dims[0], YSIXE = dims[1]
TVSCL, img
LOADCT, 12

Input the data of each of the contour paths into IDLanROI objects:

FOR I = 0, (N_ELEMENTS (PathInfo) - 1 ) DO BEGIN & $

Note
The & after BEGIN and the $ allow you to use the FOR/DO loop at the IDL
command line. These & and $ symbols are not required when the FOR/DO
loop in placed in an IDL program as shown in Programbef ineROI .pro in
the examples/doc/image subdirectory of the IDL installation directory.

Programmatically Defining ROIs Image Processing in IDL

9. Initialize oROI with the contour information of the current region:

line = [LINDGEN (PathInfo(I).N), 0] & $

OROI = OBJ_NEW('IDLanROI', $
(pathXY (*, pathInfo(I).OFFSET + line)) [0, *], $
(pathXY (*, pathInfo(I).OFFSET + line)) [1, *]) & $

10. Draw the ROI object in a Direct Graphics window using DRAW_ROI:
DRAW ROI, oROI, COLOR = 80 & $

11. Usethe IDLanROI::ComputeMask function in conjunction with
IMAGE_STATISTICS to obtain maskArea, the number of pixels covered by
theregion when it is displayed. The variable, maskResult, isinput as the value
of MASK in the second statement in order to return the maskArea:

maskResult = oROI -> ComputeMask( $
DIMENSIONS = [dims[0], dims[1]]) & $

IMAGE STATISTICS, img, MASK = maskResult, $
COUNT = maskArea & $

12. Usethe IDLanROI::ComputeGeometry function to return the geometric area
and perimeter of each region. In the following example, SPATIAL_SCALE
defines that each pixel represents 1.2 by 1.2 millimeters:

ROIStats = oOROI -> ComputeGeometry( $

AREA = geomArea, PERIMETER = perimeter, $
SPATIAL SCALE = [1.2, 1.2, 1.0]) & $

Note
The value for SPATIAL _SCALE in the previous statement is used only as
an example. The actual spatial scale value istypicaly known based upon
equipment used to gather the data.

13. Print the statistics for each ROl when it is displayed and wait 3 seconds before
proceeding to the display and analysis of the next region:

PRINT, ' ' & $

PRINT, 'Region''s mask area = '8
FIX (maskArea), ' pixels' & $

PRINT, 'Region''s geometric area = ', 8
FIX(geomArea), ' mm' & $

PRINT, 'Region''s perimeter = ', S
FIX (perimeter),' mm' & $

WAIT, 3

14. Remove each unneeded object reference after displaying the region:

OBJ_DESTROY, OROI & $

15. End the FOR loop:

ENDFOR

The outlines of the ROIs recorded by the CONTOUR function have been
trandated into ROI objects and displayed using DRAW_ROI. Each region’s
“mask area,” (computed using IDLanROI::ComputeMask in conjunction with
IMAGE_STATISTICS) shows the number of pixels covered by the region
when it is displayed on the screen.

Each region’s geometric area and perimeter, (computed using
IDLanROI::ComputeGeometry’s SPATIAL_SCALE keyword) resultsin the
following geometric area and perimeter measurements in millimeters.

Region's magk area = 2600 pixelz Region's magk area = 1669 pixelz
Region's geametric area = 3520 mm Region's geametric area = 2262 mm
Region's perimeter = 416 mm Region's perimeter = 285 mm

Region's mask area = 4193 pixelz
Region's geometric area = 5754 mm
Region's perimeter = BEE mm

Region's mask area = 26 pixels
Region's geometric area = 23 mm
Region's perimeter = 23 mm

Figure 6-7: Display of Programmatically Defined Regions

Growing a Region

The REGION_GROW function is an analysis routine that allows you to identify a
complicated region without having to manually draw intricate boundaries. This
function expands a given region based upon the constraints imposed by either a
threshold range (minimum and maximum pixel values) or by a multiplier of the
standard deviation of the original region. REGION_GROW expands an original
region to include al connected neighboring pixelsthat fall within the specified limits.

The following example interactively defines an initial region within a cross-section
of ahuman skull. Theinitial region is then expanded using both methods of region
expansion, thresholding and standard deviation multiplication. Complete the
following steps for a detailed description of the process.

Example Code
See regiongrowex.pro inthe examples/doc/image subdirectory of the IDL
installation directory for code that duplicates this example.

1. Preparethe display device and load a grayscale color table:

DEVICE, DECOMPOSED = 0, RETAIN = 2
LOADCT, O

2. Select thefile, read in the data and get the image dimensions:

file = FILEPATH('mdl107g8a.jpg', $
SUBDIRECTORY = ['examples',6 'data'])

READ JPEG, file, img, /GRAYSCALE

dims = SIZE(img, /DIMENSIONS)

3. Double the size of the image for display purposes and compute the new
dimensions:

img = REBIN(BYTSCL(img), dims[0]*2, dims[1]*2)
dims = 2*dims

4. Create awindow and display the original image:

WINDOW, 0, XSIZE = dims[0], YSIZE = dims[1], S
TITLE = 'Click on Image to Select Point of ROI'
TVSCL, img

Image Processing in IDL Growing a Region

The following figure shows the initial image.

Figure 6-8: Original Image Showing Region to be Selected

5. Definethe original region pixels. Using the CURSOR function, select the
original region by positioning your cursor over the image and clicking on the
region indicated in the previous figure by the “+” symbol. Then create a 10 by
10 square ROI, named roipixels, at the selected x, y, coordinates:

CURSOR, xi, yi, /DEVICE

x = LINDGEN(10*10) MOD 10 + xi
y = LINDGEN(10%*10) / 10 + yi
roiPixels = x + y * dims[0]

Note
A region can also be defined and grown using the XROI utility. See the

XROI procedurein the IDL Reference Guide for more information.

6. Delete the window after selecting the point:
WDELETE, O
7. Set the topmost color table entry to red:

topClr = !D.TABLE SIZE - 1
TVLCT, 255, 0, 0, topClr

Growing a Region Image Processing in IDL

8. Display theinitial region using the previously defined color:

regionPts = BYTSCL(img, TOP = (topClr - 1))

regionPts [roiPixels] = topClr

WINDOW, 0, XSIZE = dims[0], YSIZE = dims[1], $
TITLE = 'Original Region'

TV, regionPts

The following figure shows the initial ROI that will be input and expanded
with the REGION_GROW function.

Original
Region

Figure 6-9: Square ROI at Selected Coordinates

9. Using the REGION_GROW function syntax,

Result = REGION GROW(Array, ROIPixels [, /ALL_NEIGHBORS]
[, STDDEV_MULTIPLIER=value | THRESHOLD=[min,max]] )

input the original region, roipixels, and expand the region to include all
connected pixels which fall within the specified THRESHOLD range:

REGION GROW (img, roiPixels, $
[215,255])

newROIPixels
THRESHOLD

Note
If neither the THRESHOLD nor the STDDEV_MULTIPLIER keywords are

specified, REGION_GROW automatically applies THRESHOLD, using the
minimum and maximum pixels values occurring within the original region.

Image Processing in IDL Growing a Region

10. Show the results of growing the original region using threshold values:

Note
An error message such asAttempt to subscript REGIONIMG with
NEWROIPIXELS is out of range indicatesthat the pixel valueswithin
the defined region fall outside of the minimum and maximum THRESHOLD
values. Either define aregion containing pixel values that occur within the
threshold range or alter the minimum and maximum values.

regionImg = BYTSCL(img, TOP = (topClr-1))

regionImg [newROIPixels] = topClr

WINDOW, 2, XSIZE = dims[0], YSIZE = dims[1], $
TITLE = 'THRESHOLD Grown Region'

TV, regionlmg

The left-hand image in the following figure shows that the region has been
expanded to clearly identify the optic nerves. Now expand the original region
by specifying a standard deviation multiplier value as described in the
following step.

11. Expand the original region using avalue of 7 for STDDEV_MULTIPLIER:

stddevPixels = REGION GROW(img, roiPixels, $
STDDEV_MULTIPLIER = 7)

12. Create a new window and show the resulting ROI:

Growing a Region

WINDOW, 3, XSIZE = dims[0], YSIZE = dims[1], $

TITLE = "STDDEV_MULTIPLIER Grown Region"
regionImg2 = BYTSCL(img, TOP = (topClr - 1))
regionImg2 [stddevPixels] = topClr

TV, regionImg2

Image Processing in IDL

The following figure displays the results of growing the original region using
thresholding (left) and standard deviation multiplication (right).

Figure 6-10: Regions Expanded Using REGION_GROW

Note
Your results for the right-hand image may differ. Results of growing aregion using
a standard deviation multiplier will vary according to the exact mean and deviation
of the pixel values within the original region.

Creating and Displaying an ROI Mask

The IDLanROI::ComputeMask function method defines a 2D mask of aregion
object, returning an array inwhich al pixelsthat lie outside of the region have avalue
of 0. The mask can then be used to extract the portion of the original image that lies
within the ROI. The following example defines an ROI, computes a mask, appliesthe
mask to retain only the portion of the image defined by the ROI, and produces a
magnified view of the ROI. Complete the following steps for adetailed description of
the process.

Example Code
See scalemask object.prointheexamples/doc/image subdirectory of the

IDL installation directory for code that duplicates this example.

1. Select thefile, read in the data and get the image dimensions:

file = FILEPATH('md5290fcl.jpg', $
SUBDIRECTORY = ['examples',6 'data'])

READ JPEG, file, img, /GRAYSCALE

dims = SIZE(img, /DIMENSIONS)

2. Passtheimage to XROI and use the Draw Polygon tool to define the region:

XROI, img, REGIONS OUT = ROIout, /BLOCK

L] i| B[ n] x|9| o] 4| — Draw Polygon

Figure 6-11: ROI Definition in XROI

3.

10.

Close the XROI window to save the region object data in the variable, ROlout.
Assign the ROI datato the arrays, x and y:

ROIout -> GetProperty, DATA = ROIdata
x = ROIdatal0, *]
y = ROIdatal[l, *]

Set the properties of the ROI:
ROIout -> SetProperty, COLOR = [255,255,255], THICK = 2
Initialize an IDLgrlmage object containing the original image data:

oImg = OBJ NEW('IDLgrImage',6 img,$
DIMENSIONS = dims)

Create awindow in which to display the image and the ROI:

oWindow = OBJ NEW('IDLgrWindow', DIMENSIONS = dims, $
RETAIN = 2, TITLE = 'Selected ROI')

Create the view plane and initialize the view:

viewRect = [0, 0, dims[0], dims[1]]
oView = OBJ NEW('IDLgrView',6 VIEWPLANE RECT = viewRect)

Initialize amodel object and add the image and ROI to the model. Add the
model to the view and draw the view in the window to display the ROI
overlaid onto the original image:

oModel = OBJ NEW ('IDLgrModel!')
oModel -> Add, oImg

oModel -> Add, ROIout

oView -> Add, oModel

oWindow -> Draw, oView

Use the IDLanROI::ComputeM ask function to create a 2D mask of the region.
Pixels that fall outside of the ROI will be assigned a value of O:

maskResult = ROIout -> ComputeMask (DIMENSIONS = dims)

Usethe IMAGE_STATISTICS procedure to compute the area of the mask,
inputting maskResult as the MASK value. Print count to view the number of
pixels occurring within the masked region:

IMAGE STATISTICS, img, MASK = MaskResult, COUNT = count
PRINT, 'area of mask = ', count,' pixels'

Image Processing in IDL Creating and Displaying an ROl Mask

Note
The COUNT keyword to IMAGE_STATISTICS returns the number of pixels
covered by the ROl when it is displayed, the same value as that shown in the
“# Pixels’ field of XROI’s ROI Information dial og.

11. From the ROI mask, create a binary mask, consisting of only zeros and ones.
Multiply the binary mask times the original image to retain only the portion of
the image that was defined in the original ROI:

mask = (maskResult GT 0)
maskImg = img * mask

12. Using the minimum and maximum values of the ROI array, create a cropped
array, croplmg, and get its dimensions:

cropImg = maskImg[min (x) :max(x), min(y): max(y)]
cropDims = SIZE(cropImg, /DIMENSIONS)

13. Initialize an image object with the cropped region data:

oMaskImg = OBJ NEW('IDLgrImage', croplImg, $
DIMENSIONS = dims)

14. Using the cropped region dimensions, create an offset window. Multiply the x
and y dimensions times the value by which you wish to magnify the ROI:

oMaskWindow = OBJ NEW ('IDLgrWindow', $
DIMENSIONS = 2 * cropDims, RETAIN
TITLE = 'Magnified ROI', LOCATION

2 ! $

dims)

15. Create the display objects and display the cropped and magnified ROI:

oMaskView = OBJ NEW('IDLgrView',6 VIEWPLANE RECT = viewRect)
oMaskModel = OBJ NEW('IDLgrModel"')

oMaskModel -> Add, oMaskImg

oMaskView -> Add, oMaskModel

OMaskWindow -> Draw, oMaskView

The original and the magnified view of the ROI are shown in the following
figure.

Figure 6-12: Original and Magnified View of the ROI

16. Clean up object references that are not destroyed by the window manager
when you close the Object Graphics displays:

OBJ_DESTROY, [oView, oMaskView, ROIout]

Testing an ROI for Point Containment

The IDLanROI::ContainsPoints function method determines whether a point having
given coordinates lies inside, outside, on the boundary of, or on the vertex of a
designated ROI. The following example allows the creation of an ROI within an
image of the world using XROI. After exiting XROI, apoint is selected and tested to
determine its relationship to the ROI. The example then creates textual and graphical
displays of the results. Complete the following steps for a detailed description of the
process.

Example Code
See containmenttest.